

# Язык предикатного программирования P

Д.Ю. Першин, Н.С. Карнаухов, В.И. Шелехов

3 декабря 2013 г.

## Изменения версии 0.12

1. Изменены описания классов. Теперь они в конце разд. 6. Переработан разд. 10, определяющий описания процессов. Все изменения отмечены тремя плюсами +++
2. Разд. 4. Аксиомы, леммы и теоремы определяются конструкцией УТВЕРЖДЕНИЕ, одной из альтернатив ОПИСАНИЯ.
3. Разд. 3.3. В позиции вызова предиката должно быть ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ, а не ВЫРАЖЕНИЕ
4. Разд.6. Введено ВТОРИЧНОЕ-ВЫРАЖЕНИЕ в целях ограничить ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ в конструкциях типа “поле структуры”.

## 1 Введение

Язык предикатного программирования P (Predicate programming language) является универсальным и может использоваться для разработки широкого класса программ-функций. По меньшей мере, этот класс включает программы, разрабатываемые обычно на языке ФОРТРАН, для задач вычислительной математики. В целях спецификации и реализации программ из класса программ-процессов, в частности, реактивных систем, язык P расширен средствами для описания процессов, передачи сообщений и порождения процессов, работающих параллельно с процессом-родителем.

По сложившейся классификации язык P следует отнести к классу языков функционального программирования. Язык P сочетает функциональный и операторный (предикатный) стили записи алгоритмов и обладает существенно большей выразительностью по сравнению с чисто функциональными языками. В этом смысле язык P ближе к языкам логического программирования. Однако в отличие от логических языков процесс вычисления программы на языке P реализуется не логическим выводом, а явным исполнением, характерным для императивных и функциональных языков.

Язык P содержит средства спецификации программы. Программа состоит из набора определений предикатов. Определение предиката имеет вид:

$$A(x: y) \text{ pre } P(x)\{ S \} \text{ post } Q(x, y) ,$$

где A - имя определяемого предиката, x - аргументы, y - результаты, S - оператор, P(x) - предусловие и Q(x, y) - постусловие. Тотальная корректность предиката A определяется истинностью следующей формулы:

$$P(x) \Rightarrow [L(S)(x, y) \Rightarrow Q(x, y)] \& \exists y.L(S)(x, y) \quad (1)$$

где L(S) - логика оператора S - сильнейший предикат, истинный при завершении исполнения оператора S [6]. Корректность программы относительно спецификации определяется корректностью каждого определяемого предиката программы. Формулы корректности генерируются по программе автоматически, что позволяет применять для их доказательства системы автоматического доказательства, например такие, как PVS [7]. Отметим, что для императивных языков генерация формул корректности является нетривиальной задачей; кроме того, там намного сложнее исходная спецификация, генерируемые формулы корректности и их последующее доказательство.

Функциональный язык удобнее императивного для разработки алгоритма. Однако получение эффективной программы для функциональных языков проблематично даже с применением изоэтрной оптимизации в процессе трансляции. Поэтому после отладки программы на функциональном языке для достижения требуемой эффективности ее приходится переписывать на императивный язык. Преимущество технологии предикатного программирования в том, что она является сквозной: к предикатной программе

применяется набор трансформаций с получением эффективной программы на императивном расширении языка P (см. разд. 11), после чего программа может конвертироваться на любой из императивных языков: C, C++, ФОРТРАН и др. Базовыми трансформациями являются:

- замена хвостовой рекурсии циклом;
- подстановка определения предиката на место его вызова;
- склеивание переменных, реализующее замену нескольких переменных одной;
- кодирование структурных объектов низкоуровневыми структурами с использованием массивов и указателей.

C помощью трансформаций можно получить программу предельной эффективности.

Язык предикатного программирования P впервые представлен в работе [1]. Полное описание языка P впервые дано в препринте [2]. Последующие модификации и расширения языка инициированы использованием языка для описания различных алгоритмов [3-5]. Введены средства спецификации определений предикатов в виде предусловий и постусловий. Имеется опыт использования языка P в качестве языка публикации алгоритмов. Язык P расширен для спецификации и реализации реактивных систем, определяемых в виде совокупности взаимодействующих процессов [5]. Введены средства изображения процессов, операторы приема и отправки сообщений, средства динамического порождения процессов.

Все предыдущие версии языка P по стилю были ближе к языкам Паскаль и Модула-2. Последняя версия языка P, представленная в настоящем описании, по синтаксису, набору операторов и операций и др. особенностям существенно приближена к стилю языков типа C. Версия 0.6 описана в препринте [12]. Как следствие, язык P станет более комфортным для программирующих на языках C, C++, Java, C#. Изменения синтаксиса языка P стали причиной проведения серии серьезных модификаций во всем языке. Определенное влияние на новую версию языка P оказал язык спецификаций PVS [7]. Введены алгебраические типы. Последовательности заменены списками.

Синтаксис языка P описывается на расширенном языке Бэкусовских нормальных форм (БНФ) со следующими особенностями:

- терминальные символы выделены жирным шрифтом
- [ фрагмент ] - означает возможное отсутствие в синтаксическом правиле заключенного в квадратные скобки фрагмента; фрагмент определяет последовательность терминальных и нетерминальных символов
- ( фрагмент )<sup>\*</sup> - определяет повторение фрагмента нуль или более раз; круглые скобки могут быть опущены, если фрагмент состоит из одного символа
- ( фрагмент )<sup>+</sup> - определяет повторение фрагмента один или более раз
- запись вида [:CLASS:] обозначает символ указанного класса; используются следующие классы символов:

alpha	Буквенный символ, принадлежащий латинскому или русскому алфавиту; разрешаются заглавные и строчные буквы
digit	Цифра (0, 1, 2, 3, 4, 5, 6, 7, 8 или 9)
alnum	Символ, принадлежащий alpha или digit
blank	Пробел или символ табуляции
xdigit	Шестнадцатеричная цифра (digit либо заглавная или строчная буква от A до F)
print	Символ, для которого определено начертание (т.е., символ из alnum, blank либо какой-либо другой символ, который можно отобразить)

## 2 Лексемы

Текст программы представлен в виде последовательности строк символов. Переход на новую строку эквивалентен символу "пробел". Программа может содержать комментарии, текст которых считается не принадлежащим тексту программы. Комментарий начинается парой символов /\* и завершается парой символов \*/. Второй вид комментариев начинается с пары символов // и продолжается до конца текущей строки текста. Текст программы составляется из лексем следующего вида:

ЛЕКСЕМА ::=  
ИДЕНТИФИКАТОР | КЛЮЧЕВОЕ-СЛОВО | КОНСТАНТА | ОПЕРАЦИЯ | РАЗДЕЛИТЕЛЬ | МЕТКА

ИДЕНТИФИКАТОР ::= НАЧАЛО-ИДЕНТИФИКАТОРА СИМВОЛ-ИДЕНТИФИКАТОРА\*  
НАЧАЛО-ИДЕНТИФИКАТОРА ::= \_ | [:alpha:]  
СИМВОЛ-ИДЕНТИФИКАТОРА ::= \_ | [:alnum:]

Идентификатор используется для именованя предикатов, переменных, типов и других объектов. Использование специального имени “\_” в качестве результирующей переменной вызова предиката (см. разд. 3.3) обозначает пустой результат, неиспользуемый в дальнейшем вычислении. Имя “\_” зарезервировано в языке P - его нельзя использовать для именованя объектов программы. Другие идентификаторы, начинающиеся с “\_”, считаются обычными и могут использоваться в описаниях программы.

МЕТКА ::= СИМВОЛ-ИДЕНТИФИКАТОРА\*  
КЛЮЧЕВОЕ-СЛОВО ::=  
after | array | as | axiom | bool | break | case | char | class | else | enum  
| exists | extends | default | false | for | forall | formula | if | in  
| inf | int | import | lemma | message | module | nan | nat | new | nil | or  
| post | pre | process | predicate | real | receive | send | set | spawn | string  
| struct | subtype | switch | theorem | type | true | union | var | while | with | xor

ОПЕРАЦИЯ ::=  
+ | - | \* | / | % | ^ | ! | << | >> | ~ | & | | | ? | = | < | > | <=  
| >= | != | => | <=>

РАЗДЕЛИТЕЛЬ ::=  
( | ) | [ | ] | { | } | , | ; | : | . | .. | [:blank:]

КОНСТАНТА ::=  
ЦЕЛАЯ-КОНСТАНТА  
| ВЕЩЕСТВЕННАЯ-КОНСТАНТА  
| СИМВОЛЬНАЯ-КОНСТАНТА  
| СТРОКОВАЯ-КОНСТАНТА  
| ЛОГИЧЕСКАЯ-КОНСТАНТА  
| КОНСТАНТА-ПУСТО

ЦЕЛАЯ-КОНСТАНТА ::= ЦЕЛОЕ-ЧИСЛО  
ЦЕЛОЕ-ЧИСЛО ::= [ЗНАК] ЦИФРЫ | 0x ШЕСТНАДЦАТЕРИЧНЫЕ-ЦИФРЫ  
ЗНАК ::= + | -  
ЦИФРЫ ::= ЦИФРА+  
ЦИФРА ::= [:digit:]  
ШЕСТНАДЦАТЕРИЧНЫЕ-ЦИФРЫ ::= ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА+  
ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА ::= [:xdigit:]  
blank  
ВЕЩЕСТВЕННАЯ-КОНСТАНТА ::=  
[ЗНАК] ЦИФРЫ [. ЦИФРЫ] [ПОРЯДОК]  
| [ЗНАК] **inf**  
| **nan**  
ПОРЯДОК ::= (e | E) [ЗНАК] ЦИФРЫ

Для вещественных типов (см. разд. 7) определены специальные значения: **inf** (бесконечность) и **nan** (не число). Значение **nan** возникает в процессе вычисления как результат взятия квадратного корня из отрицательного числа, деления ноля на ноль, умножения ноля на бесконечность и других операций над вещественными числами, когда результат не может быть определен.

СИМВОЛЬНАЯ-КОНСТАНТА ::= ' СИМВОЛ '  
СИМВОЛ ::= ОБЫЧНЫЙ-СИМВОЛ | СПЕЦИАЛЬНЫЙ-СИМВОЛ | ШЕСТНАДЦАТЕРИЧНЫЙ-КОД-СИМВОЛА  
ОБЫЧНЫЙ-СИМВОЛ ::= любой символ из [:print:], кроме ' и "  
СПЕЦИАЛЬНЫЙ-СИМВОЛ ::= \" | \' | \\ | \0 | \n | \r | \t  
ШЕСТНАДЦАТЕРИЧНЫЙ-КОД-СИМВОЛА ::= \x ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА  
[ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА [ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА [ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА]]]

Константа типа **char** (см. разд. 7) может быть либо символом класса **print** (см. разд. 1), либо специальной комбинацией символов, заключённой в одинарные кавычки. Специальные комбинации символов могут также встречаться в составе строковых констант.

Таблица 1: Специальные комбинации символов

<code>\''</code>	Двойные кавычки
<code>\'</code>	Одинарные кавычки
<code>\\</code>	Обратная косая черта
<code>\0</code>	Символ с кодом 0
<code>\n</code>	Перевод строки (код 10)
<code>\r</code>	Возврат каретки (код 13)
<code>\t</code>	Символ табуляции
<code>\xNNNN</code>	Символ с шестнадцатеричным кодом NNNN, состоящим не более чем из четырёх цифр.

```

СТРОКОВАЯ-КОНСТАНТА ::= " СИМВОЛ * "
ЛОГИЧЕСКАЯ-КОНСТАНТА ::= true | false
КОНСТАНТА-ПУСТО ::= nil

```

### 3 Предикаты

Программа состоит из набора определений предикатов.

#### 3.1 Определение предиката

```

ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА ::= ИМЯ-ПРЕДИКАТА ОПИСАНИЕ-ПРЕДИКАТА
ИМЯ-ПРЕДИКАТА ::= ИДЕНТИФИКАТОР

```

Значением ОПИСАНИЯ-ПРЕДИКАТА является предикат, обозначаемый именем предиката.

```

ОПИСАНИЕ-ПРЕДИКАТА ::= ОПИСАНИЕ-ПРЕДИКАТА-ФУНКЦИИ |
                        ОПИСАНИЕ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ
ОПИСАНИЕ-ПРЕДИКАТА-ФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА [pre ПРЕДУСЛОВИЕ] ТЕЛО-ПРЕДИКАТА [post ПОСТУСЛОВИЕ]
ЗАГОЛОВОК-ПРЕДИКАТА ::= ( [ОПИСАНИЯ-АРГУМЕНТОВ]: ОПИСАНИЯ-РЕЗУЛЬТАТОВ )
ТЕЛО-ПРЕДИКАТА ::= БЛОК
ПРЕДУСЛОВИЕ ::= ФОРМУЛА
ПОСТУСЛОВИЕ ::= ФОРМУЛА

```

Значения аргументов предиката должны удовлетворять предусловию. По завершении исполнения тела предиката должно быть истинным постусловие, связывающее значения аргументов и результатов.

```

ОПИСАНИЯ-РЕЗУЛЬТАТОВ ::=
    ИЗОБРАЖЕНИЕ-ТИПА ИМЯ-РЕЗУЛЬТАТА (, ИМЯ-РЕЗУЛЬТАТА)*
    [, ОПИСАНИЯ-РЕЗУЛЬТАТОВ]
ИМЯ-РЕЗУЛЬТАТА ::= ИДЕНТИФИКАТОР | ИДЕНТИФИКАТОР '

```

Имя вида *имя'* используется для именованного результирующего параметра, при условии что *имя* описано в качестве одного из аргументов с тем же типом. В императивной программе, получаемой трансформацией предикатной программы, результат с именем *имя'* склеивается с соответствующим аргументом *имя*. Если тип аргумента *имя* параметризован (см. разд. 7), то значение параметра для типа результата *имя'* может отличаться.

```

ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ ::=
    ИЗОБРАЖЕНИЕ-ТИПА ИМЯ-АРГУМЕНТА (, ИМЯ-АРГУМЕНТА)* |
    ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА
ОПИСАНИЯ-АРГУМЕНТОВ ::= ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ [, ОПИСАНИЯ-АРГУМЕНТОВ]
ИМЯ-АРГУМЕНТА ::= ИДЕНТИФИКАТОР

```

Listing 1: Примеры определений предикатов

```

1 assign (int from : int to) { to = from }
2 main (list (string) argv : int ret_code) { ret_code = 0 }
3 power (real x, int p : real x') { x' = x^p }

```

```

ОПИСНИЕ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ]
    ТЕЛО-ПРЕДИКАТА [ПОСТУСЛОВИЯ-ВЕТВЕЙ]
ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ( [ОПИСАНИЯ-АРГУМЕНТОВ] : ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ )
ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ ::=
    [ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ] [# МЕТКА-ВЕТВИ]
    (: [ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ] [# МЕТКА-ВЕТВИ] )+
ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ ::= ОПИСАНИЯ-РЕЗУЛЬТАТОВ
МЕТКА-ВЕТВИ ::= МЕТКА

```

После двоеточия описываются параметры-результаты очередной ветви гиперфункции. Ветви гиперфункции именуется метками. При отсутствии метки при описании результатов ветви, ветвь именуется целым - порядковым номером ветви начиная с 1. Исполнение гиперфункции завершается выбором одной из ветвей и вычислением значений результатов этой ветви. При этом результаты других ветвей не определены. Оператор перехода #имяВетви в теле предиката завершает исполнение гиперфункции ветвью с именем имяВетви.

```

ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ ::= [ОБЩЕЕ-ПРЕДУСЛОВИЕ] (ПРЕДУСЛОВИЕ-ВЕТВИ)+
ОБЩЕЕ-ПРЕДУСЛОВИЕ ::= pre ПРЕДУСЛОВИЕ
ПРЕДУСЛОВИЕ-ВЕТВИ ::= pre МЕТКА-ВЕТВИ : ПРЕДУСЛОВИЕ

```

Отсутствие общего предусловия определяет допустимость произвольных значений аргументов в соответствии с их типами. Предусловие ветви определяет дополнительное условие на значения аргументов, при котором исполнение гиперфункции завершается выбором этой ветви. При этом значения аргументов должны удовлетворять общему предусловию. Предусловие последней ветви обычно не указывается, поскольку оно получается дополнением предусловий предыдущих ветвей.

```

ПОСТУСЛОВИЯ-ВЕТВЕЙ ::= (ПОСТУСЛОВИЕ-ВЕТВИ)+
ПОСТУСЛОВИЕ-ВЕТВИ ::= post МЕТКА-ВЕТВИ : ПОСТУСЛОВИЕ

```

Постусловие ветви связывает значения аргументов и результатов ветви. Ветвь без результатов постусловия не имеет.

### 3.2 Спецификация предиката

Предикат, используемый в программе, должен иметь определение предиката. Если предикат определяется в другом модуле программы, то предикат может быть представлен своей спецификацией.

```

СПЕЦИФИКАЦИЯ-ПРЕДИКАТА ::= ИМЯ-ПРЕДИКАТА ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА

```

Имя предиката обозначает предикат, представленный описанием спецификации предиката.

```

ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА ::= ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ФУНКЦИИ |
    ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ
ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА [pre ПРЕДУСЛОВИЕ] [post ПОСТУСЛОВИЕ]
ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ] [ПОСТУСЛОВИЯ-ВЕТВЕЙ]

```

### 3.3 Вызов предиката

Элементарным оператором программы является вызов предиката.

```

ВЫЗОВ-ПРЕДИКАТА ::= ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ | ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ
ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ ::= ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] : РЕЗУЛЬТАТЫ )
ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ::=
    [ИМЯ-МОДУЛЯ .] ИМЯ-ПРЕДИКАТА | [ОБЪЕКТ-КЛАССА .] ИМЯ-ПРЕДИКАТА |
    ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ

```

ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ должно быть предикатного типа; его значением является предикат, запускаемый на исполнение данным вызовом.

```

ОБЪЕКТ-КЛАССА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ

```

Доступ к предикату, являющемуся методом класса, реализуется через объект класса.

```

АРГУМЕНТЫ ::= СПИСОК-ВЫРАЖЕНИЙ
СПИСОК-ВЫРАЖЕНИЙ ::= ВЫРАЖЕНИЕ [, СПИСОК-ВЫРАЖЕНИЙ]

```

Типы аргументов вызова должны быть совместимы (см. разд. 7) с типами соответствующих аргументов определения (или спецификации) вызываемого предиката.

```

РЕЗУЛЬТАТЫ ::= ПЕРЕМЕННАЯ [, РЕЗУЛЬТАТЫ] |
                РЕЗУЛЬТАТ-ЛОКАЛ [, РЕЗУЛЬТАТЫ]
РЕЗУЛЬТАТ-ЛОКАЛ ::= ОПИСАНИЕ-ПЕРЕМЕННОЙ
ОПИСАНИЕ-ПЕРЕМЕННОЙ ::= ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ИДЕНТИФИКАТОР
ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ::= ИЗОБРАЖЕНИЕ-ТИПА [:blank:] | var [:blank:]

```

Использование `var` возможно вместо соответствующего ИЗОБРАЖЕНИЕ-ТИПА, поскольку тип результата-локала в большинстве случаев можно восстановить по определению вызываемого предиката. Описатель `var` также может использоваться при описании локальной переменной в случае, когда тип этой переменной легко определяется из контекста дальнейшего использования переменной (см. разд. 4, 5).

Типы результатов вызова должны совпадать с типами соответствующих результатов в определении вызываемого предиката. Переменная, представленная описанием результата-локала, определяется как локальная в теле предиката, содержащем данный вызов. Описание локальной результирующей переменной внутри вызова эквивалентно описанию этой переменной перед вызовом.

Использование специального имени “`_`” в качестве результирующей переменной обозначает пустой результат: результат по этой позиции, получаемый при завершении исполнения вызова предиката, не используется в дальнейшем исполнении. Имя “`_`” зарезервировано в языке P - его нельзя использовать для именованя объектов программы.

```

ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] (: РЕЗУЛЬТАТЫ-ВЕТВИ)+ )
РЕЗУЛЬТАТЫ-ВЕТВИ ::= [РЕЗУЛЬТАТЫ] [ОПЕРАТОР-ПЕРЕХОДА]
ОПЕРАТОР-ПЕРЕХОДА ::= # МЕТКА

```

В качестве метки в операторе указывается либо метка ветви предиката, в теле которого находится данный вызов, либо метка ветви обработчика; подробнее см. в разд. 5.

Исполнение вызова предиката реализуется следующим образом. Результатом исполнения аргументов вызова является набор значений. Вычисление каждого аргумента вызова реализуется независимо от вычисления других, т.е. параллельно. Полученный набор значений аргументов присваивается соответствующим входным параметрам определения вызываемого предиката. Далее исполняется тело определения вызываемого предиката. В процессе исполнения тела определяется итоговая ветвь предиката и вычисляются значения результирующих переменных по этой ветви. Наконец, полученные значения результирующих переменных присваиваются соответствующим результирующим переменным вызова предиката, и исполнение вызова завершается по этой ветви вызова. Если в данной ветви вызова указан оператор перехода, то либо происходит переход на локальную метку, либо завершается тело предиката, содержащего вызов, той ветвью, которая указана в операторе перехода. Отметим, что подстановка аргументов и результатов предиката реализуется по значению.

```

ВЫЗОВ-ФУНКЦИИ ::= ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] )

```

Вызов предиката-функции может иметь вид вызова функции. Результатом исполнения является набор значений результирующих переменных исполненного определения предиката.

Listing 2: Примеры вызова предикатов, определения, спецификации и вызова гиперфункций

```

1 real r;
2 sqrt (2 : r);
3 real q = sqrt (3);
4
5 Comp (list(int) s: : int d, list (int) r) // спецификация гиперфункции Comp
6 pre 1: s = nil
7 post 2: s = cons(d, r) ;
8 /* если список s — пуст, реализуется первая ветвь гиперфункции, иначе реализуется
9    вторая ветвь с результатами: d — первый элемент, r — хвост списка */
10
11 elemTwo (list(int) s: int e #value : #empty)
12 pre empty: s = nil or s.cdr = nil
13 {
14   Comp (s : #empty : int e1, list(int) s1 #ok)

```

```

15     case ok: Comp (s1 : #empty : e, list (int) s2 #value);
16 }
17 post value: e = (s.cdr).car;
18 /* гиперфункция elemTwo извлекает второй элемент списка, если элемент существует
*/

```

В приведенном выше определении предиката `elemTwo` переход по локальной метке `ok` является излишним. Кроме того, результаты `e1` и `s2` далее нигде не используются. Определение предиката `elemTwo`, представленное ниже, исправляет отмеченные недостатки:

```

1 elemTwo (list (int) s: int e #value : #empty)
2 pre 1: s = nil or s.cdr = nil
3 {   Comp (s : #empty : _, list (int) s1 );
4     Comp (s1 : #empty : e, _ #value);
5 }
6 post value: e = s.cdr.car;

```

## 4 Программа

Программа состоит из одного или нескольких модулей. Модуль определяет независимую часть программы.

```

ОПИСАНИЕ-МОДУЛЯ ::= [ЗАГОЛОВОК-МОДУЛЯ] ОПИСАНИЯ-МОДУЛЯ
ЗАГОЛОВОК-МОДУЛЯ ::= module ИМЯ-МОДУЛЯ [ ( ОПИСАНИЯ-АРГУМЕНТОВ ) ];
ИМЯ-МОДУЛЯ ::= ИДЕНТИФИКАТОР

```

Аргументы, описанные в круглых скобках, являются формальными параметрами модуля. Структура ОПИСАНИЙ-АРГУМЕНТОВ определена в разд. 3.1.

```

ОПИСАНИЯ-МОДУЛЯ ::= ОПИСАНИЕ-МОДУЛЯ [ ; ОПИСАНИЯ-МОДУЛЯ ]      +++
ОПИСАНИЕ-МОДУЛЯ ::= ИМПОРТ-МОДУЛЯ | ОПИСАНИЕ                     +++
ОПИСАНИЕ ::=                                                         +++
    ОПИСАНИЕ-ТИПА | ОПИСАНИЕ-ПЕРЕМЕННЫХ |
    ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА | СПЕЦИФИКАЦИЯ-ПРЕДИКАТА |
    ОПИСАНИЕ-ФОРМУЛЫ | УТВЕРЖДЕНИЕ |
    ОПРЕДЕЛЕНИЕ-КЛАССА | ОПИСАНИЕ-СООБЩЕНИЯ | ОПРЕДЕЛЕНИЕ-ПРОЦЕССА

```

ОПИСАНИЕ-ФОРМУЛЫ (см. разд. 9) определяет функцию или предикат; они встречаются в предусловиях, постусловиях и формулах. УТВЕРЖДЕНИЕ определяет аксиому, лемму или теорему (см. разд. 9).

ОПРЕДЕЛЕНИЕ-КЛАССА описано в конце разд.6. Две последних альтернативы ОПИСАНИЯ используются для задания процессов; см. разд. 10.

```

ИМПОРТ-МОДУЛЯ ::= import ИМЯ-МОДУЛЯ [ ( АРГУМЕНТЫ ) ] [ as ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ ]
ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ ::= ИДЕНТИФИКАТОР

```

Имена модулей, встречающиеся в описаниях данного модуля, должны быть определены конструкцией ИМПОРТ-МОДУЛЯ. Если импортируемый модуль имеет формальные параметры, то в круглых скобках задается соответствующий набор фактических параметров. Конструкция АРГУМЕНТЫ определена в разд. 3.3. Правила подстановки фактических параметров на место формальных те же самые, что и для вызова предиката. ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ используется в теле модуля как эквивалент ИМЯ-МОДУЛЯ ( АРГУМЕНТЫ ).

Описания переменных как часть ОПИСАНИЙ-МОДУЛЯ определяют глобальные переменные, являющиеся внешними параметрами задачи. В определениях предикатов эти переменные считаются аргументами. Они используются в теле предиката, но не упоминаются в заголовке.

-----+++

```

ОПИСАНИЕ-ПЕРЕМЕННЫХ ::=
    ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ИМЯ-ПЕРЕМЕННОЙ (, ИМЯ-ПЕРЕМЕННОЙ)*
ОПИСАНИЕ-ПЕРЕМЕННЫХ-С-ИНИЦИАЛИЗАЦИЕЙ ::=
    ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ
    ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ (, ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ)*
ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ ::= ИМЯ-ПЕРЕМЕННОЙ = ВЫРАЖЕНИЕ | ИМЯ-ПЕРЕМЕННОЙ

```

Тип выражения должен быть совместим с типом переменной, которой присваивается значение переменной.

Модуль определяет совокупность имен и обозначаемых ими объектов: предикатов, типов, переменных, полей структур, конструкторов и распознавателей объединений, процессов, классов и сообщений. Объекты, представленные описаниями модуля, определяют область глобальных имен. Каждое описание предиката определяет автономную систему локализации имен: параметров, меток и локальных переменных. В описании предиката одно имя не может использоваться для двух разных объектов. Если глобальное имя используется в определении предиката, то это имя не может определяться в качестве имени метки, локальной переменной или параметра.

Следующей областью локализации являются конструкции: изображение подтипа, определение массива, элемент агрегата вида “итерация выражений”. Переменные, определенные в каждой из этих конструкций, локализованы в конструкции и недоступны вне нее. Класс также определяет независимую область локализации. Поля класса доступны префиксацией объекта класса (см. разд. 6).

Исполнение программы начинается исполнением предиката с именем `main`. Рекомендован предикат со следующими параметрами: `main (list(string) argv : int ret_code)`. Параметр `argv` поставляет массив входных аргументов при исполнении программы. Параметру `ret_code` присваивается значение кода выхода.

Для вывода информации используется оператор `print`, аргументы которого - выражения и массивы.

## 5 Операторы

```
БЛОК ::= { ОПЕРАТОР }
ОПЕРАТОР ::= [ОПИСАНИЕ-ПЕРЕМЕННЫХ ;]{ ЗАМКНУТЫЙ-ОПЕРАТОР |
              [ОПИСАНИЕ-ПЕРЕМЕННЫХ ;] ПАРАЛЛЕЛЬНЫЙ-ОПЕРАТОР |
              [ОПИСАНИЕ-ПЕРЕМЕННЫХ ;] ОПЕРАТОР-СУПЕРПОЗИЦИИ
```

Описанные переменные считаются локальными в теле предиката. Тип локальной переменной может быть также определен описателем `var` (см. разд. 3.3) в случае, когда тип легко определяется из контекста дальнейшего использования переменной.

```
ЗАМКНУТЫЙ-ОПЕРАТОР ::=
    ОПЕРАТОР-ПРИСВАИВАНИЯ | ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ |
    ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ] |
    УСЛОВНЫЙ-ОПЕРАТОР | БЛОК [ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ] | ОПЕРАТОР-ВЫБОРА |
    ОПРЕДЕЛЕНИЕ-ЛОКАЛЬНОГО-ПРЕДИКАТА
ОПЕРАТОР-ПРИСВАИВАНИЯ ::= ПЕРЕМЕННАЯ = ВЫРАЖЕНИЕ
```

В результате исполнения оператора присваивания переменной присваивается значение выражения. Типы выражения и переменной должны быть совместимы (см. разд. 7).

```
УСЛОВНЫЙ-ОПЕРАТОР ::= ОПЕРАТОР-ЕСЛИ else ЗАМКНУТЫЙ-ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]
ОПЕРАТОР-ЕСЛИ ::= if ( ВЫРАЖЕНИЕ ) ЗАМКНУТЫЙ-ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]
```

В операторе перехода указывается либо метка ветви гиперфункции, либо метка ветви обработчика, следующего за блоком, содержащем данный условный оператор.

Listing 3: Пример использования условного оператора

```
1 if (x > 0) s = 1
2 else if (x < 0) s = -1
3     else s = 0
```

```
ПАРАЛЛЕЛЬНЫЙ-ОПЕРАТОР ::= ЗАМКНУТЫЙ-ОПЕРАТОР (|| ЗАМКНУТЫЙ-ОПЕРАТОР)+
```

Операторы, образующие параллельный оператор, должны иметь непересекающиеся наборы результирующих переменных. Исполнение параллельного оператора складывается из исполнения входящих в него операторов. Операторы выполняются независимо друг от друга; они могут исполняться параллельно.

```
ОПЕРАТОР-СУПЕРПОЗИЦИИ ::= ОПЕРАТОР (; ОПЕРАТОР)+
```

В цепочке операторов, составляющих оператор суперпозиции, каждый следующий оператор использует значения локальных переменных, присвоенных предыдущими операторами. Каждая пара соседних операторов в цепочке должна быть связана хотя бы одной локальной переменной; если такой связи нет, их композиция должна быть оформлена параллельным оператором.

```

ЗАГОЛОВОК-ОПЕРАТОРА-ВЫБОРА ::= switch ( ВЫРАЖЕНИЕ )
ОПЕРАТОР-ВЫБОРА                ::= ЗАГОЛОВОК-ОПЕРАТОРА-ВЫБОРА
                                { ОПЕРАТОР-АЛЬТЕРНАТИВЫ+
                                  [ default : ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]]
                                }
ОПЕРАТОР-АЛЬТЕРНАТИВЫ ::= case АЛЬТЕРНАТИВА ( , АЛЬТЕРНАТИВА)* : ОПЕРАТОР
                                                                [ОПЕРАТОР-ПЕРЕХОДА]
АЛЬТЕРНАТИВА                ::= ВЫРАЖЕНИЕ | ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА

```

Выполняется тот оператор альтернативы, для которого значение АЛЬТЕРНАТИВЫ совпадает со значением выражения в заголовке оператора выбора. Оператор после **default** выполняется в случае, когда нет ОПЕРАТОРА-АЛЬТЕРНАТИВЫ с требуемым значением АЛЬТЕРНАТИВЫ. Трактовка операторов перехода та же, что и для условного оператора.

Если в позиции выражения в заголовке оператора выбора находится переменная типа объединения, то в качестве АЛЬТЕРНАТИВЫ указывается ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА. Особенности выполнения ОПЕРАТОРА-АЛЬТЕРНАТИВЫ определены в разд. 7.

```

ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ ::= (case [МЕТКА-ВЕТВИ-ОБРАБОТЧИКА :] ОПЕРАТОР)+
МЕТКА-ВЕТВИ-ОБРАБОТЧИКА ::= МЕТКА

```

Оператор обработки ветвей следует за вызовом гиперфункции либо за блоком, содержащим операторы перехода. Если предыдущий оператор нормально завершается, то оператор обработки ветвей игнорируется. Нормальное завершение возможно и для вызова гиперфункции, когда на одной из ветвей гиперфункции опущена метка перехода, что означает суперпозицию (или параллельную композицию) с последующим оператором, а на другой ветви имеется переход на метку ветви обработчика или на метку ветви гиперфункции, в теле которой находится данный оператор.

Исполнение оператора обработки ветвей инициируется оператором перехода, находящемся в предыдущем операторе. При этом исполняется оператор, помеченный соответствующей меткой, после чего исполнение оператора обработки ветвей и предыдущего оператора считается завершившимся.

В вызове гиперфункции могут быть опущены переходы, а в последующем операторе обработки ветвей - метки ветвей обработчика. В этом случае порядок ветвей обработчика соответствует порядку ветвей вызова гиперфункции.

Listing 4: Пример вызова гиперфункции с обработчиком ветвей

```

1 A(i: y #1: #2)
2   case 1: B(y, i: u)
3   case 2: D(i: u);

```

```

ОПРЕДЕЛЕНИЕ-ЛОКАЛЬНОГО-ПРЕДИКАТА ::= ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА

```

Определение локального предиката допускает использование параметров и локальных переменных предиката, в теле которого это определение находится. Локальный предикат доступен лишь в теле предиката, внутри которого он описан.

## 6 Выражения

```

ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ::=
    ПЕРЕМЕННАЯ | ВЫЗОВ-ФУНКЦИИ | ( ВЫРАЖЕНИЕ ) | ПОЛЕ-ОБЪЕДИНЕНИЯ    +++
ВТОРИЧНОЕ-ВЫРАЖЕНИЕ ::=
    ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ | КОНСТАНТА | АГРЕГАТ | ИМЯ-ПРЕДИКАТА |
    ГЕНЕРАТОР-ПРЕДИКАТА | | ИМЯ-ТИПА |
    МОДИФИКАЦИЯ | ЭЛЕМЕНТ-СПИСКА | ОПРЕДЕЛЕНИЕ-МАССИВА |
    МЕТОД | СООБЩЕНИЕ |
    КОНСТРУКТОР | РАСПОЗНАВАТЕЛЬ    +++

```

Имя типа может быть аргументом вызова предиката и изображения типа. Определение ЭЛЕМЕНТА-СПИСКА дается в разд. 7. Операция ОПРЕДЕЛЕНИЕ-МАССИВА описывается в разд. 8.3. КОНСТРУКТОР, РАСПОЗНАВАТЕЛЬ и ПОЛЕ-ОБЪЕДИНЕНИЯ относятся к объектам типа объединения и описаны в разд. 7. Использование СООБЩЕНИЯ в качестве значения первичного выражения определено в разд.10.

ПЕРЕМЕННАЯ ::= ИМЯ-ПЕРЕМЕННОЙ | ЭЛЕМЕНТ-МАССИВА | ПОЛЕ-СТРУКТУРЫ |  
 ПЕРЕМЕННАЯ-КЛАССА | ВЫРЕЗКА-МАССИВА | МУЛЬТИ-ПЕРЕМЕННАЯ

ВЫРЕЗКА-МАССИВА определена в разд. 8.2.

ИМЯ-ПЕРЕМЕННОЙ ::= [ИМЯ-МОДУЛЯ .] ИДЕНТИФИКАТОР ['] +++

Переменная вида *имя'* используется для именованного результата предиката. Переменная *имя'* склеивается с переменной *имя*, являющейся аргументом предиката (см. разд. 3.1).

ЭЛЕМЕНТ-МАССИВА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ [ ИНДЕКСЫ-МАССИВА ]  
 ИНДЕКСЫ-МАССИВА ::= СПИСОК-ВЫРАЖЕНИЙ

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “массив”. Значения ИНДЕКСОВ-МАССИВА должны принадлежать типам индексов соответствующего типа массива.

ПОЛЕ-СТРУКТУРЫ ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-ПОЛЯ

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “структура”.

ИМЯ-ПОЛЯ ::= ИДЕНТИФИКАТОР  
 ПЕРЕМЕННАЯ-КЛАССА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-ПОЛЯ +++

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “класс”.

МУЛЬТИ-ПЕРЕМЕННАЯ ::= | СПИСОК-ПЕРЕМЕННЫХ | |  
 СПИСОК-ПЕРЕМЕННЫХ

Два разных способа представления мультипеременной считаются равноценными.

СПИСОК-ПЕРЕМЕННЫХ ::= ПЕРЕМЕННАЯ [, СПИСОК-ПЕРЕМЕННЫХ ]

Мульти-переменная определяет упорядоченный набор переменных. Число переменных в наборе должно быть больше 1.

АГРЕГАТ ::= АГРЕГАТ-СТРУКТУРА | АГРЕГАТ-МАССИВ | АГРЕГАТ-МНОЖЕСТВО | АГРЕГАТ-СПИСОК

Агрегат определяет значение структурного типа. Агрегат-массив определяет массив, агрегат-множество - подмножество некоторого множества - значение типа **set** (см. разд. 7), агрегат-структура - структуру (как набор полей), агрегат-список - список, заданный последовательностью элементов.

АГРЕГАТ-СТРУКТУРА ::= [ИЗОБРАЖЕНИЕ-ТИПА] ( ЭЛЕМЕНТЫ-АГРЕГАТА )  
 АГРЕГАТ-МАССИВ ::= [ИЗОБРАЖЕНИЕ-ТИПА] [ [ЭЛЕМЕНТЫ-АГРЕГАТА] ]  
 АГРЕГАТ-МНОЖЕСТВО ::= [ИЗОБРАЖЕНИЕ-ТИПА] { [ЭЛЕМЕНТЫ-АГРЕГАТА] }  
 АГРЕГАТ-СПИСОК ::= [ИЗОБРАЖЕНИЕ-ТИПА] [[ ЭЛЕМЕНТЫ-АГРЕГАТА ]]

Агрегат [ ] обозначает пустой массив (когда верхняя граница индексов меньше нижней). Агрегат { } определяет пустое множество. Если пара скобок [ ] или { } встречаются рядом при изображении АГРЕГАТ-МАССИВА, то соседние скобки должны быть разделены пробелом.

ЭЛЕМЕНТЫ-АГРЕГАТА ::= ЭЛЕМЕНТ-АГРЕГАТА [, ЭЛЕМЕНТЫ-АГРЕГАТА]

Значением агрегата является набор значений элементов агрегата, перечисленных в скобках. Агрегат может иметь иерархическую структуру, поскольку элементом агрегата может быть агрегат. Тип агрегата определяется типом позиции, в которой находится агрегат. Значение агрегата должно соответствовать типу позиции. Тип агрегата может быть указан явно ИЗОБРАЖЕНИЕМ-ТИПА - это не является необходимым, но может улучшить восприятие программы.

ЭЛЕМЕНТ-АГРЕГАТА ::= [ИНДЕКС-ЭЛЕМЕНТА-АГРЕГАТА :] ВЫРАЖЕНИЕ  
 ИНДЕКС-ЭЛЕМЕНТА-АГРЕГАТА ::= ВЫРАЖЕНИЕ | ИМЯ-ПОЛЯ

Индексы элементов, если присутствуют, должны соответствовать индексам элементов массива или именам полей структуры.

Listing 5: Пример задания агрегата

```
1 type Ar4 = array (real , 1..4);
2 Ar4 x = [ 0, 1, 2, 3];
```

ГЕНЕРАТОР-ПРЕДИКАТА ::=  
**predicate** ОПИСАНИЕ-ПРЕДИКАТА

Исполнение генератора предиката создает новый предикат, операторная часть которого определяется телом предиката, находящегося в составе ОПИСАНИЯ-ПРЕДИКАТА; см. разд. 3.1. В теле предиката фиксируются значения свободных переменных (отличных от аргументов и результатов) на момент исполнения генератора. Новый предикат становится значением генератора предиката.

МЕТОД ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-МЕТОДА +++  
ИМЯ-МЕТОДА ::= ИДЕНТИФИКАТОР

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “класс”.

УНАРНОЕ-ВЫРАЖЕНИЕ ::= УНАРНАЯ-ОПЕРАЦИЯ ВТОРИЧНОЕ-ВЫРАЖЕНИЕ +++  
УНАРНАЯ-ОПЕРАЦИЯ ::= + | - | ! | ~  
БИНАРНОЕ-ВЫРАЖЕНИЕ ::= ВЫРАЖЕНИЕ БИНАРНАЯ-ОПЕРАЦИЯ ВЫРАЖЕНИЕ  
БИНАРНАЯ-ОПЕРАЦИЯ ::= \* | / | % | + | - | << | >> | in |  
< | > | <= | >= | = | != | & | ^ | or | xor | => | <=>

Семантика операций и их приоритеты определены ниже в таблице. Типы аргументов каждой операции должны соответствовать операции. Операция “+” для объединения массивов описана в разд. 8.4.

МОДИФИКАЦИЯ ::= МОДИФИКАЦИЯ-СТРУКТУРЫ | МОДИФИКАЦИЯ-МАССИВА  
МОДИФИКАЦИЯ-СТРУКТУРЫ ::= ВЫРАЖЕНИЕ with АГРЕГАТ  
МОДИФИКАЦИЯ-МАССИВА ::= ВЫРАЖЕНИЕ with АГРЕГАТ |  
ВЫРАЖЕНИЕ with ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ

В структурном значении выражения, обозначающего массив или структуру, меняются компоненты, представленные АГРЕГАТОМ или ОПРЕДЕЛЕНИЕМ-МАССИВА-ПО-ЧАСТЯМ (см. разд. 8.3). Значением операции является обновленный массив или структура.

Listing 6: Примеры модификации массивов

```
1 type ar0_4 = array (int , 0..4);
2 ar0_4 ones = [ 1, 1, 1, 2, 1 ];
3 ar0_4 ones1 = ones [3: 1]; // [ 1, 1, 1, 1, 1 ]
4 type Vec(nat n) = array (real , 1..n);
5 perm(nat n, Vec(n) b, nat m, k : Vec(n) b')
6 { b' = b with [k: b [m], m: b [k]] };

```

УСЛОВНОЕ-ВЫРАЖЕНИЕ ::= ВЫРАЖЕНИЕ ? ВЫРАЖЕНИЕ : ВЫРАЖЕНИЕ

Первое выражение имеет тип “логический”. Если первое ВЫРАЖЕНИЕ завершается идентификатором, то идентификатор и последующий символ “?” должны быть разделены пробелом.

МУЛЬТИ-ВЫРАЖЕНИЕ ::= | СПИСОК-ВЫРАЖЕНИЙ | |  
СПИСОК-ВЫРАЖЕНИЙ

Значением мульти-выражения является набор значений, получающихся в результате вычисления перечисленных выражений. Выражения вычисляются независимо, т.е. параллельно. Мульти-выражение может находиться в правой части оператора присваивания, если в левой находится мульти-переменная.

ВЫРАЖЕНИЕ ::= ВТОРИЧНОЕ-ВЫРАЖЕНИЕ | УНАРНОЕ-ВЫРАЖЕНИЕ |  
БИНАРНОЕ-ВЫРАЖЕНИЕ | УСЛОВНОЕ-ВЫРАЖЕНИЕ | МУЛЬТИ-ВЫРАЖЕНИЕ

Таблица 2: Операции в порядке уменьшения приоритета

$\wedge$	Возведение в степень
$+$ , $-$ , $!$ , $\sim$	Унарные плюс и минус, логическое отрицание и поэлементное дополнение для множеств, побитовое дополнение для ограниченных целых
$*$ , $/$ , $\%$	Арифметическое умножение, деление и остаток от целочисленного деления
$+$ , $-$	Арифметическое сложение и вычитание; объединение и вычитание множеств; конкатенация списков; объединение массивов с непересекающимися типами индексов
$\ll$ , $\gg$	Побитовый (поэлементный) сдвиг влево и вправо для целых
<b>in</b>	Проверка вхождения элемента в множество
$<$ , $>$ , $\leq$ , $\geq$	Операции арифметического сравнения
$=$ , $\neq$	Проверка на равенство и неравенство
<b>&amp;</b>	Пересечение множеств, логическая конъюнкция, побитовое И для ограниченных целых
<b>xor</b>	Симметрическая разность для множеств, исключительное ИЛИ для логических выражений и ограниченных целых (побитовое)
<b>or</b>	Объединение множеств, логическая дизъюнкция, побитовое ИЛИ для ограниченных целых
$\Rightarrow$	Импликация
$\Leftrightarrow$	Логическое тождество
<b>?:</b>	Трёхместная условная операция

## 7 Типы

Всякая переменная имеет некоторый тип. Тип может быть атрибутом языковой конструкции. Семантика конструкции налагает определенные ограничения на значения типов. Для каждой подконструкции некоторой конструкции определяется *позиция* ее вхождения внутри конструкции. Это, например, позиции операндов операций, позиции правой и левой частей в операторе присваивания, позиция фактического параметра в вызове предиката и др. При наличии в конструкции двух позиций типы этих позиций должны быть *согласованы*. Частным случаем согласованности является *совместимость* типов. Например, тип выражения правой части оператора присваивания должен быть совместим с типом переменной левой части оператора присваивания, но не наоборот. Реализуется неявное преобразование (называемое *приведением*) значения совместимого типа к типу, требуемому позицией.

```

ОПИСАНИЕ-ТИПА ::= type ИМЯ-ТИПА [( ПАРАМЕТРЫ-ТИПА )] = ИЗОБРАЖЕНИЕ-ТИПА |
                ПРЕДОПИСАНИЕ-ТИПА
ИМЯ-ТИПА      ::= ИДЕНТИФИКАТОР
ПАРАМЕТРЫ-ТИПА ::= ИЗОБРАЖЕНИЕ-ТИПА [:blank:] ИДЕНТИФИКАТОР (, ИДЕНТИФИКАТОР)*
                [, ПАРАМЕТРЫ-ТИПА]

```

Описание типа связывает имя типа с его изображением. Тип может быть параметризован, т.е. зависеть от набора значений переменных и типов, указанных в качестве параметров.

```

ПРЕДОПИСАНИЕ-ТИПА ::= type ИМЯ-ТИПА

```

Если имя типа используется до его описания, что возможно для рекурсивных определений типов, то перед первым вхождением имени оно должно быть декларировано с помощью предописания.

```

ИЗОБРАЖЕНИЕ-ТИПА ::=
    ИЗОБРАЖЕНИЕ-ПРИМИТИВНОГО-ТИПА | ИЗОБРАЖЕНИЕ-ТИПА-ПО-ИМЕНИ |
    ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА | ИЗОБРАЖЕНИЕ-ТИПА-ПРЕДИКАТА |
    ИЗОБРАЖЕНИЕ-ПОДТИПА | ИЗОБРАЖЕНИЕ-ПЕРЕЧИСЛЕНИЯ | ИЗОБРАЖЕНИЕ-СТРУКТУРНОГО-ТИПА |
    ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ | ИЗОБРАЖЕНИЕ-КЛАССА
ИЗОБРАЖЕНИЕ-ПРИМИТИВНОГО-ТИПА ::=
    nat [(РАЗЯДНОСТЬ-ЦЕЛЫХ)]
    | int [(РАЗЯДНОСТЬ-ЦЕЛЫХ)]
    | real [(РАЗЯДНОСТЬ-ВЕЩЕСТВЕННЫХ)]
    | bool
    | char

```

Тип `nat` является подмножеством неотрицательных чисел типа `int`, `char` – множество символов некоторого алфавита. Конкретные представления примитивных типов даются в императивном расширении языка (см. разд. 11).

Разрядность определяет максимальное число битов в представлении значений типа. Указание разрядности в изображении примитивного типа является частью императивного расширения языка.

Значения типа `nat` совместимы с типом `int`, а `int` – с типом `real`. Значения типа меньшей размерности совместимы с аналогичными типами большей размерности.

**ИЗОБРАЖЕНИЕ-ТИПА-ПО-ИМЕНИ ::=**  
**[ИМЯ-МОДУЛЯ .] ИМЯ-ТИПА [( АРГУМЕНТЫ )]**

Ранее определенный тип может быть представлен своим именем. Изображение параметризованного типа представляется с конкретными параметрами.

**ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА ::= `type` ИМЯ-ТИПА**

Объявление типа в качестве параметра описания типа или определения предиката реализуется описателем `type`. В качестве обозначения типа может использоваться выражение типа `type`, которым, например, может являться параметр предиката или параметризованного типа. Например:

Listing 7: Пример использования типов в качестве параметров

```

1 type Point (type T) = struct (T x, y);
2 type Polyline (type T) = list (Point (T));
3
4 decimate (type T, list (T) data : list (T) data') {
5   data' = (data = nil) ?
6     nil :
7     data.car + (len(data) < 3 ? nil : decimate(T, data.cdr.cdr))
8 };
9
10 reverse (type T, list (T) data : list (T) data') {
11   data' = (data = nil) ? nil : reverse (T, data.cdr) + data.car
12 };
13
14 Polyline (real) line = [[ (0.0, 0.0), (0.1, 0.2), (0.2, 0.5),
15   (0.3, 0.6), (0.4, 0.3), (0.5, 0.0) ]];
16
17 Polyline (real) line1 = reverse (decimate (real, line));
18 // [[ (0.4, 0.3), (0.2, 0.5), (0.0, 0.0) ]]

```

В примере выше идентификатор `T` использован в качестве типа-параметра, а тип `real` – как фактический параметр.

```

1 squareVec (type T, list (T) data : list (T) data') {
2   data' = (data = nil) ? nil : data.car ^ 2 + squareVec (T, data.cdr)
3 };
4
5 list (int) squares = squareVec (int, [[ 0, 1, 2, 3, 4 ]]); // [[ 0, 1, 4, 9, 16 ]]

```

Выше приведен еще один пример использования типа `T` в качестве параметра предиката.

**ИЗОБРАЖЕНИЕ-ТИПА-ПРЕДИКАТА ::=**  
**`predicate` ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА**

Имя параметра предиката может быть опущено. В этом случае описывается только тип параметра. Изображение типа предиката используется для описания переменной предикатного типа.

Listing 8: Пример использования предикатного типа

```

1 type int_list = list (int);
2 map_list (int_list src, predicate (int : int) func : int_list dest) {
3   dest = (src = nil) ? nil : func (src.car) + map_list (src.cdr, func)
4 };
5 int_list numbers = [[ 0, 1, 2, 3, 4, 5 ]];
6 int_list squares = map_list (numbers, predicate (int a : int b) { b = a * a; });
7 // squares = [[ 0, 1, 4, 9, 16, 25 ]]

```

```

ИЗОБРАЖЕНИЕ-ПОДТИПА ::=
    subtype ( ОПИСАНИЕ-ПЕРЕМЕННОЙ : ВЫРАЖЕНИЕ ) |
    ИЗОБРАЖЕНИЕ-ДИАПАЗОНА

```

Конструкция `subtype(T x: выражение)` определяет подмножество типа `T`. Описание переменной `x` действует в пределах этой конструкции. Множество значений переменной `x`, для которых логическое `ВЫРАЖЕНИЕ` имеет значение “истина”, является определяемым подтипом типа `T`. Если `ВЫРАЖЕНИЕ` содержит другие переменные, то все они являются параметрами изображаемого типа. Конструкция `ОПИСАНИЕ-ПЕРЕМЕННОЙ` определена в разд. 3.3.

```

ИЗОБРАЖЕНИЕ-ДИАПАЗОНА ::= ВЫРАЖЕНИЕ .. ВЫРАЖЕНИЕ

```

Изображение диапазона является подмножеством типа `int`, `enum`, или `char`. Изображение типа `subtype(int i: i>=1 & i<=n+1)` эквивалентно изображению диапазона: `1..n+1`.

```

1 type Odd = subtype (int n: n % 2 = 1);
2 type diap10_20 = 10..20;

```

```

ИЗОБРАЖЕНИЕ-СТРУКТУРНОГО-ТИПА ::=
    ИЗОБРАЖЕНИЕ-ТИПА-СТРУКТУРЫ | ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ |
    ИЗОБРАЖЕНИЕ-ТИПА-СПИСКА | ИЗОБРАЖЕНИЕ-ТИПА-МНОЖЕСТВА | ИЗОБРАЖЕНИЕ-ТИПА-МАССИВА
ИЗОБРАЖЕНИЕ-ТИПА-СТРУКТУРЫ ::= struct ( ОПИСАНИЕ-ПОЛЕЙ )
ОПИСАНИЕ-ПОЛЕЙ ::= ИЗОБРАЖЕНИЕ-ТИПА СПИСОК-ИМЕН-ПОЛЕЙ
    [, ОПИСАНИЕ-ПОЛЕЙ]
СПИСОК-ИМЕН-ПОЛЕЙ ::= ИМЯ-ПОЛЯ [, СПИСОК-ИМЕН-ПОЛЕЙ]

```

Значение типа структуры состоит из совокупности значений полей структуры. Имена полей должны быть различны. Число полей должно быть не менее двух.

Listing 9: Пример определения структуры

```

1 type Point = struct ( int x, y );
2 Point pt = (10, 20);

```

```

ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ ::= union ( ОПИСАНИЯ-КОНСТРУКТОРОВ )
ОПИСАНИЯ-КОНСТРУКТОРОВ ::= ОПИСАНИЕ-КОНСТРУКТОРА (, ОПИСАНИЕ-КОНСТРУКТОРА)*

```

Значением типа объединения является значение одного из конструкторов, перечисленных в списке описаний конструкторов. Число конструкторов должно быть не меньше двух.

```

ОПИСАНИЕ-КОНСТРУКТОРА ::= ИМЯ-КОНСТРУКТОРА [ ( ОПИСАНИЕ-ПОЛЕЙ ) ]
ИМЯ-КОНСТРУКТОРА ::= ИДЕНТИФИКАТОР

```

Поля, определяемые в круглых скобках, являются полями объединения. Значения этих полей используются в качестве аргументов конструктора. Имена полей объединения по всем конструкторам должны быть различны.

Работа с объектами типа объединения реализуется с помощью следующих конструкций: конструктор, распознаватель и поле объединения. Все они являются **ПЕРВИЧНЫМИ-ВЫРАЖЕНИЯМИ** (см. разд. 6).

```

КОНСТРУКТОР ::= ИМЯ-КОНСТРУКТОРА [ ( СПИСОК-ВЫРАЖЕНИЙ ) ]

```

Выражения, подставляемые в качестве аргументов `КОНСТРУКТОРА`, по их числу и типам должны соответствовать `ОПИСАНИЯМ-ПОЛЕЙ` в соответствующем `ОПИСАНИИ-КОНСТРУКТОРА`. Правила соответствия такие же, как для вызова предиката. Значением конструкции `КОНСТРУКТОР` является имя конструктора вместе со значениями его аргументов, если аргументы присутствуют.

```

РАСПОЗНАВАТЕЛЬ ::= ИМЯ-РАСПОЗНАВАТЕЛЯ ( ВЫРАЖЕНИЕ )
ИМЯ-РАСПОЗНАВАТЕЛЯ ::= ИМЯ-КОНСТРУКТОРА ?

```

Символ “?” является частью имени распознавателя и пишется слитно с именем конструктора, без пробела. `ВЫРАЖЕНИЕ` имеет тип объединения. Распознаватель является функцией типа `bool`. Значение `РАСПОЗНАВАТЕЛЯ` истинно, если значением `ВЫРАЖЕНИЯ` является конструктор с именем `ИМЯ-КОНСТРУКТОРА`.

```

ПОЛЕ-ОБЪЕДИНЕНИЯ ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-ПОЛЯ

```

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “объединение”. Значением ПОЛЯ-ОБЪЕДИНЕНИЯ является значение поля с именем ИМЯ-ПОЛЯ в структуре переменной типа “объединение”. Использование конструкции ПОЛЕ-ОБЪЕДИНЕНИЯ корректно лишь в случае истинности распознавателя ИМЯ-КОНСТРУКТОРА? ( ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ) для конструктора, к которому относится поле с именем ИМЯ-ПОЛЯ, в противном случае исполнение конструкции не определено.

Имена конструкторов и распознавателей являются глобально определенными в теле модуля, содержащего ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ.

```
ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА ::= ИМЯ-КОНСТРУКТОРА [ ( ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА ) ]
ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА ::= ОПРЕДЕЛЕНИЕ-ПОЛЯ-КОНСТРУКТОРА
                                     [, ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА]
ОПРЕДЕЛЕНИЕ-ПОЛЯ-КОНСТРУКТОРА ::= [ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ] ИДЕНТИФИКАТОР
```

Конструкция ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА используется в качестве АЛЬТЕРНАТИВЫ после case в операторе выбора (см. разд. 5) для выражения типа объединения. Значением АЛЬТЕРНАТИВЫ является конструктор с именем ИМЯ-КОНСТРУКТОРА и набором переменных, обозначающих поля данного конструктора. Эти переменные являются локальными в ОПЕРАТОРЕ-АЛЬТЕРНАТИВЫ, исполняемом для данной АЛЬТЕРНАТИВЫ. Типы переменных могут не указываться, поскольку они определяются из ИЗОБРАЖЕНИЯ-ТИПА-ОБЪЕДИНЕНИЯ.

Listing 10: Пример использования объединений

```
1 type int_tree = union (
2   leaf (int val),
3   node (int_tree lhs, int v, int_tree rhs)
4 );
5
6 int_tree foo;
7
8 // ...
9
10 switch (foo) {
11   case leaf(v0): print("leaf", v0)
12   case node(l, v1, r): print("node with ", v1)
13 };
14
15 int_tree one_leaf = leaf(42);
16 int_tree small_tree = node(leaf(1), 2, node(leaf(3), 4, leaf(5)));
17 // small_tree задаёт следующее дерево:
18 //      2
19 //     / \
20 //    1  4
21 //     / \
22 //    3  5
```

```
ИЗОБРАЖЕНИЕ-ПЕРЕЧИСЛЕНИЯ ::= enum ( ИДЕНТИФИКАТОР (, ИДЕНТИФИКАТОР)* )
```

Тип перечисления является частным случаем типа объединения с конструкторами без аргументов. Описатель enum трактуется как эквивалент union.

Listing 11: Пример перечисления

```
type fruit = enum (apple, orange, banana);
```

Структура вида “список” представляется как упорядоченная совокупность однородных элементов. Тип списка из элементов типа T определяется следующим описанием:

Listing 12: Определение списка

```
1 type list (type T) = union (
2   nil,
3   cons (T car, list(T) cdr)
4 );
```

Тип `list` считается определенным в языке Р и не требует описания в программе. Имена `list`, `nil` и `cons` глобально определены в предикатной программе.

Конструктор `nil` определяет пустой список, не содержащий элементов. Остальные значения типа `list` представляются конструктором `cons(x, s)`, где элемент `x` есть первый элемент списка - голова списка, а `s` есть оставшаяся часть списка - хвост списка. Функция `len(s)` определяет число элементов списка `s`. Операция конкатенации `s1 + s2` определяет список, состоящий из элементов списка `s1`, за которыми следуют элементы списка `s2`. Аргументами конкатенации могут быть также элементы списка. Функция `last(s)` определяет последний элемент непустого списка `s`; функция `prec(s)` - оставшуюся часть списка без последнего элемента.

**ЭЛЕМЕНТ-СПИСКА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ [ ИНДЕКС-ЭЛЕМЕНТА ]**

Данная конструкция определяет значение элемента с указанным индексом в списке. Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является список.

**ИНДЕКС-ЭЛЕМЕНТА ::= ВЫРАЖЕНИЕ**

Циклы `for` и `while` часто используются для представления циклов, реализуемых с помощью операторов перехода, в целях улучшения структуры программы.

Для списка `s` конструкция `s[i]` определяет `i`-ый элемент. Допустимыми являются значения индекса `i` от 0 до `len(s)-1`.

Listing 13: Примеры операций со списком

```

1 type int_list = list (int);
2 int_list s    = [[ 1, 1, 2, 3, 5, 8 ]];
3 int s_car     = s.car;           // s_car = 1
4 int_list s_cdr = s.cdr;         // s_cdr = [ 1, 2, 3, 5, 8 ]
5 int_list ss   = s + [[ 13, 21 ]]; // ss = [ 1, 1, 2, 3, 5, 8, 13, 21 ]
6 int count    = len (s);         // count = 6
7 int e        = s[3];            // e = 3

```

**СТРОКОВЫЙ-ТИП ::= string**

Строковый тип `string` является обозначением типа списка `list(char)`. В дополнение к определенным выше операциям со списками имеется возможность задания конкретного значения строкового типа в виде строковой константы (см. разд. 2).

**ИЗОБРАЖЕНИЕ-ТИПА-МНОЖЕСТВА ::= set ( ИЗОБРАЖЕНИЕ-БАЗОВОГО-ТИПА )**

**ИЗОБРАЖЕНИЕ-БАЗОВОГО-ТИПА ::= ИЗОБРАЖЕНИЕ-ТИПА**

Тип множества есть множество всех подмножеств некоторого конечного базового типа. Имеется унарная операция `~` дополнения множества. Определены: операция `+` объединения множеств, операция `-` разности множеств, а также вычитания элемента из множества, операция `&` пересечения множеств, операция `or` объединения множеств, операция `in` принадлежности элемента множеству (см. разд. 6). Число элементов в множестве реализуется функцией `len`.

Операция `mask` преобразует множество в целое значение. Операция `bits` реализует обратную операцию преобразования целого в множество.

Listing 14: Пример использования множеств

```

1 type int_set_t = set (1..1000);
2 int_set_t pow_2 = { 0, 2, 4, 8, 16, 32 };
3 bool is_pow_2   = 4 in pow_2; // is_pow_2 = true
4 int_set_t more_pow_2 = pow_2 + { 64, 128 };

ИЗОБРАЖЕНИЕ-КЛАССА ::= class [extends ИМЯ-СУПЕРКЛАССА] { ОПИСАНИЯ-КЛАССА }      ***
ОПРЕДЕЛЕНИЕ-КЛАССА ::= class ИМЯ-КЛАССА [( ПАРАМЕТРЫ-ТИПА )]
                        [extends ИМЯ-СУПЕРКЛАССА] { ОПИСАНИЯ-КЛАССА }

ИМЯ-КЛАССА ::= ИМЯ-ТИПА
ИМЯ-СУПЕРКЛАССА ::= ИМЯ-ТИПА

```

ОПРЕДЕЛЕНИЕ-КЛАССА эквивалентно следующему описанию типа:

**type ИМЯ-КЛАССА [( ПАРАМЕТРЫ-ТИПА )] = ИЗОБРАЖЕНИЕ-КЛАССА**

Класс представляет независимую область локализации имен, определяемых ОПИСАНИЯМИ-КЛАССА.

```
ОПИСАНИЯ-КЛАССА ::= ОПИСАНИЕ-КЛАССА [ ; ОПИСАНИЯ-КЛАССА ]
ОПИСАНИЕ-КЛАССА ::= ОПИСАНИЕ | ОПИСАНИЕ-КОНСТРУКТОРА-КЛАССА
```

Элементы класса - поля и методы класса. Поля класса представлены переменными, определяемыми описаниями переменных внутри ОПИСАНИЙ-КЛАССА. Методы класса - предикаты, представленные определениями и спецификациями предикатов внутри ОПИСАНИЙ-КЛАССА. Методами класса являются также определенные в классе процессы.

При наличии подконструкции **extends** в описании класса элементы класса определяются как продолжение элементов суперкласса.

Объект класса - значение переменной типа класса. Значением объекта является набор полей класса аналогично значению структурного типа.

Доступ к элементу класса реализуется через объект класса с помощью конструкции ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ.имя-элемента, где объект является значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ. Доступ к элементу класса внутри класса осуществляется непосредственно по имени элемента. Для обозначения объекта, определяемого данным классом, используется имя **this**.

```
ОПИСАНИЕ-КОНСТРУКТОРА-КЛАССА ::=
    ИМЯ-КЛАССА ([ОПИСАНИЯ-АРГУМЕНТОВ]) { ТЕЛО-КОНСТРУКТОРА }
ТЕЛО-КОНСТРУКТОРА ::= [ ОПЕРАТОР ]
```

Создание нового объекта класса реализуется через вызов конструктора, значением которого является созданный объект. Значения полей класса формируются в теле конструктора, а также начальной инициализацией полей при их описании. Если для вызова ИМЯ-КЛАССА( ) нет соответствующего конструктора с пустым списком аргументов, то такой конструктор считается определенным и его тело пусто. В результате работы конструктора все поля должны быть сформированы. Конструктор класса содержит явный или неявный вызов соответствующего конструктора для суперкласса.

## 8 Массивы

### 8.1 Описание типа массива

```
ИЗОБРАЖЕНИЕ-ТИПА-МАССИВА ::= array ( ИЗОБРАЖЕНИЕ-ТИПА-ЭЛЕМЕНТА , ИЗМЕРЕНИЯ-МАССИВА )
ИЗОБРАЖЕНИЕ-ТИПА-ЭЛЕМЕНТА ::= ИЗОБРАЖЕНИЕ-ТИПА
ИЗМЕРЕНИЯ-МАССИВА ::= ИЗОБРАЖЕНИЕ-ТИПА-ИНДЕКСА [ , ИЗМЕРЕНИЯ-МАССИВА ]
ИЗОБРАЖЕНИЕ-ТИПА-ИНДЕКСА ::= ИЗОБРАЖЕНИЕ-ТИПА
```

Значение массива состоит из совокупности элементов. Каждый элемент доступен по набору индексов в массиве: для массива **A** элемент с набором индексов **m** есть **A[m]**. Число индексов совпадает с числом измерений массива. Тип каждого индекса определяет совокупность допустимых значений индекса и должен быть конечным.

Имеется операция “+” для объединения двух массивов-операндов в один массив при условии, что типы элементов совпадают, а множества наборов индексов операндов совместимы по типам и не пересекаются.

Listing 15: Пример описаний массивов и их инициализации

```
1 type int_vec = array (int , 1..5);
2 type int_mtx = array (int , 1..2 , 1..3);
3
4 int_vec v = [ 3 , 1 , 4 , 1 , 5 ];
5 int_vec w = for (x) x*x;
6 int v_2 = v [2]; // v_2 = 1
7
8 int_mtx m = [ [ 1 , 2 , 3 ] , [ 104 , 5 , 6 ] ];
9 int e = m [2 , 3]; // e = 6
```

### 8.2 Вырезка массива

```
ВЫРЕЗКА-МАССИВА ::= ВЫРАЖЕНИЕ-МАССИВ [ СУЖЕННЫЙ-НАБОР-ТИПОВ-ИНДЕКСОВ ]
ВЫРАЖЕНИЕ-МАССИВ ::= ВЫРАЖЕНИЕ
СУЖЕННЫЙ-НАБОР-ТИПОВ-ИНДЕКСОВ ::= НАБОР-ИНДЕКСОВ
```

```

НАБОР-ИНДЕКСОВ ::= ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ [ , НАБОР-ИНДЕКСОВ ]
ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ ::= ЗНАЧЕНИЕ-ИНДЕКСА | ИЗОБРАЖЕНИЕ-ДИАПАЗОНА
ЗНАЧЕНИЕ-ИНДЕКСА ::= ВЫРАЖЕНИЕ

```

Вырезка массива определяет массив как часть некоторого массива заданием подмножества наборов индексов. Достаточно задать сужение хотя бы по одному измерению массива. В частности, допустимо сужение к единственному значению индекса. Однако вырезка массива не может превратиться в единственный элемент массива. Результатом исполнения вырезки массива является переменная типа “массив”. Нового массива при этом не создается.

### 8.3 Определение массива

```

ОПРЕДЕЛЕНИЕ-МАССИВА ::= ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА |
                          АГРЕГАТ-МАССИВ |
                          ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ
ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ ::= ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА

```

Результатом вычисления ОПРЕДЕЛЕНИЯ-МАССИВА является значение массива. Вычисление реализуется итерацией по всевозможным значениям набора индексов массива. Для каждого набора индексов вычисляется соответствующий элемент массива. Вычисление значений разных элементов может проводиться параллельно.

```

ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА ::= ВЫРАЖЕНИЕ

```

ВЫРАЖЕНИЕ для элемента массива зависит от индексов, указанных в ОПРЕДЕЛЕНИИ-ИНДЕКСОВ.

```

ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ::= for ( ЗАДАНИЕ-ИНДЕКСОВ )
ЗАДАНИЕ-ИНДЕКСОВ      ::= ОПРЕДЕЛЕНИЕ-ИНДЕКСА [ , ЗАДАНИЕ-ИНДЕКСОВ ]
ОПРЕДЕЛЕНИЕ-ИНДЕКСА ::= [ ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ] ИНДЕКС
ИНДЕКС                ::= ИДЕНТИФИКАТОР

```

Переменные, обозначающие индексы, локальны в ОПРЕДЕЛЕНИИ-МАССИВА. При определении индекса обычно используется описатель `var` (см. разд. 4). Указание типа индекса требуется в редких случаях, когда тип массива (в том числе и типы индексов массива) трудно определить из позиции, в которой находится ОПРЕДЕЛЕНИЕ-МАССИВА. Отметим, что в ОПРЕДЕЛЕНИИ-ИНДЕКСА тип индекса, если он задан явно, должен точно покрывать множество значений индекса; обычно, это диапазон типа `int`.

Listing 16: Примеры определения массивов

```

1 type ar1_5 = array (int , 1..5);
2 ar1_5 squ;
3 squ = for (var i) i*i;
4 ar1_5 r = for (var i) 100 - i;

```

```

ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА ::=
  { (ОПРЕДЕЛЕНИЕ-ЧАСТИ-МАССИВА)+ [default : ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА ] }
ОПРЕДЕЛЕНИЕ-ЧАСТИ-МАССИВА ::= case ИНДЕКСЫ-ЧАСТИ : ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА
ИНДЕКСЫ-ЧАСТИ      ::= ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ | ( НАБОР-ИНДЕКСОВ ) |
                      ( ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ [ , ИНДЕКСЫ-ЧАСТИ ] ) |
                      ( ( НАБОР-ИНДЕКСОВ ) [ , ИНДЕКСЫ-ЧАСТИ ] )

```

ИНДЕКСЫ-ЧАСТИ определяют некоторое подмножество на произведении типов индексов. Эти подмножества не должны пересекаться для разных ОПРЕДЕЛЕНИЙ-ЧАСТИ-МАССИВА. Определение элементов массива для наборов индексов, не принадлежащих ни одной из указанных частей массива, реализуется частью `default`. При отсутствии части `default` объединение подмножеств наборов индексов для разных частей должно совпадать с полным множеством наборов индексов. Вычисление элементов массива по каждой из частей массива проводится независимо, возможно параллельно.

Если ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА используется в операции модификации массива (см. разд. 6), то часть `default` отсутствует.

Listing 17: Пример определения массива по частям

```

1 type Ar(nat k) = array (real , 1..k);
2 F (nat n, Ar(n) x: Ar(n+1) x')
3 { x' = for (var j) { case 1..n : x[j] + 1 case n + 1 : 0 } }

```

Listing 18: Пример перестановки двух строк матрицы

```

1 type MATR(nat k) = array(real, 1..k, 1..k);
2 perm_lines(nat n, MATR(n) a, nat k, m : MATR(n) a' )
3 pre 1 <= k < m <= n
4 {
5     a' = a with for (var i, j) {
6         case (k, 1..n): a[m, j]
7         case (m, 1..n): a[k, j]
8     }

```

В матрице **a** переставляются места строки с номерами **k** и **m**. Перестановка реализуется применением операции модификации (см. разд. 6) двух строк в массиве **a**; остальные строки остаются неизменными. Задание нового значения двух строк матрицы реализуется конструкцией ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА.

## 8.4 Объединение массивов

Операндами операции “+” объединения массивов:

ВЫРАЖЕНИЕ-МАССИВ + ВЫРАЖЕНИЕ-МАССИВ

являются выражения, значениями которых являются массивы. Объединяемые массивы должны иметь совпадающие типы элементов и непересекающиеся типы индексов, причем типы индексов должны принадлежать некоторому общему типу. Результатом операции является массив. Тип индексов результирующего массива есть объединение типов индексов операндов, а тип элементов тот же, что и у операндов. Произвольный элемент результирующего массива есть либо элемент первого массива-операнда, либо элемент второго.

## 9 Формулы

Формулы, используемые в качестве предусловий и постусловий предикатов, есть формулы типизированного исчисления предикатов высших порядков. Подформула, входящая в предусловие или постусловие, может быть определена отдельно в виде описания формулы.

ОПИСАНИЕ-ФОРМУЛЫ ::=

**formula** ИМЯ-ФОРМУЛЫ ( ОПИСАНИЯ-АРГУМЕНТОВ [: ТИП-РЕЗУЛЬТАТА] ) = ФОРМУЛА

ИМЯ-ФОРМУЛЫ ::= ИДЕНТИФИКАТОР

ТИП-РЕЗУЛЬТАТА ::= ИЗОБРАЖЕНИЕ-ТИПА

ОПИСАНИЕ-ФОРМУЛЫ вводит имя формулы для обозначения выражения, представленного ФОРМУЛОЙ. Переменные, входящие в ФОРМУЛУ, должны быть указаны в ОПИСАНИЯХ-АРГУМЕНТОВ; см. разд. 3.1. Описание формулы помещается среди описаний модуля программы; см. разд. 4. ФОРМУЛА является выражением типа ТИП-РЕЗУЛЬТАТА. Если ТИП-РЕЗУЛЬТАТА не указан, то ФОРМУЛА имеет тип **bool**.

Использование формулы, определенной ОПИСАНИЕМ-ФОРМУЛЫ, в других формулах реализуется через вызов формулы.

ВЫЗОВ-ФОРМУЛЫ ::= ИМЯ-ФОРМУЛЫ ( СПИСОК-ВЫРАЖЕНИЙ )

Описание формулы может быть рекурсивным: ФОРМУЛА в правой части описания может содержать рекурсивный вызов этой формулы. Не допускается взаимной рекурсии в описаниях двух разных формул. Другое требование: рекурсивный вызов должен находиться в позитивной позиции ФОРМУЛЫ. Понятия позитивной и негативной позиции определены ниже.

ФОРМУЛА ::= ВЫРАЖЕНИЕ | ( ФОРМУЛА ) | ВЫЗОВ-ФОРМУЛЫ |  
! ФОРМУЛА | ФОРМУЛА & ФОРМУЛА | ФОРМУЛА **or** ФОРМУЛА |  
ФОРМУЛА => ФОРМУЛА | ФОРМУЛА <=> ФОРМУЛА | КВАНТОРНАЯ-ЧАСТЬ ФОРМУЛА

Все альтернативы приведенного правила, кроме первых трех, определяют формулы типа **bool**, т.е. предикаты. Операции “!”, “&” и “or” имеют тот же смысл, что и для логических выражений. Операция “=>” обозначает импликацию, “<=>” - логическое тождество.

КВАНТОРНАЯ-ЧАСТЬ ::= КВАНТОР СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ . [КВАНТОРНАЯ-ЧАСТЬ]

КВАНТОР ::= КВАНТОР-ВСЕОБЩНОСТИ | КВАНТОР-СУЩЕСТВОВАНИЯ

КВАНТОР-ВСЕОБЩНОСТИ ::= ! | **forall**

КВАНТОР-СУЩЕСТВОВАНИЯ ::= ? | **exists**

Использование `forall` вместо “!” предпочтительно при вхождении квантора внутри формулы, поскольку “!” ассоциируется также с отрицанием.

```
СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ ::=
    ИЗОБРАЖЕНИЕ-ТИПА ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ (, ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ) *
    [, СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ ]
ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ ::= ИДЕНТИФИКАТОР
```

Приоритеты логических связок в порядке убывания следующие: `!`, `&`, `=>`, `<=>`, кванторы `!`, `forall`, `?` и `exists`.

Определим понятие позиции, позитивной или негативной, для произвольного вхождения подформулы. Позиция ФОРМУЛЫ в качестве правой части ОПИСАНИЯ-ФОРМУЛЫ является позитивной. Допустим, формула X есть одна из следующих формул: `A & B`, `A or B`, `!C`, `C => A`, `!y.A`, `?y.A`. Для перечисленных случаев подформулы A и B имеют ту же позицию, что формула X, а подформула C меняет позицию на противоположную. Все остальные позиции подформул считаются неизвестными.

```
УТВЕРЖДЕНИЕ ::= [ИМЯ-УТВЕЖДЕНИЯ :] ОПИСАТЕЛЬ-УТВЕРЖДЕНИЯ ФОРМУЛА
ИМЯ-УТВЕЖДЕНИЯ ::= ИДЕНТИФИКАТОР
ОПИСАТЕЛЬ-УТВЕРЖДЕНИЯ ::= axiom | lemma | theorem
```

ИМЯ-УТВЕЖДЕНИЯ используется в системе автоматического доказательства для идентификации утверждения. Утверждения с описателем `axiom` считаются априори истинными. Утверждения с описателем `lemma` или `theorem` должны быть доказаны в системе автоматического доказательства.

## 10 Процессы

+++

Предикатные программы соответствуют классу программ-функций для задач дискретной и вычислительной математики. Во многих приложениях, в частности, при создании систем управления, где программа представлена в виде набора параллельных взаимодействующих процессов. Примерами таких программ являются протоколы и телекоммуникационные системы. В целях спецификации и реализации данного класса программ язык P расширен средствами для описания процессов, передачи сообщений и порождения процессов, работающих параллельно с процессом-родителем.

```
ОПРЕДЕЛЕНИЕ-ПРОЦЕССА ::=
    process ИМЯ-ПРОЦЕССА [( [ОПИСАНИЯ-АРГУМЕНТОВ] :
        ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ )]
    { СОСТОЯНИЕ-ПРОЦЕССА СПЕЦИФИКАЦИЯ-ПРОЦЕССА ТЕЛО-ПРОЦЕССА }
ИМЯ-ПРОЦЕССА ::= ИДЕНТИФИКАТОР
```

Процесс может быть бесконечным. Если процесс может завершиться, то завершение реализуется по одному или нескольким разным выходам процесса. Результаты возможного завершающегося процесса определяются так же как у гиперфункции.

```
СОСТОЯНИЕ-ПРОЦЕССА ::= ОПИСАНИЕ-ПЕРЕМЕННЫХ
```

Состояние процесса определяется набором переменных, модифицируемых при исполнении процесса.

```
ТЕЛО-ПРОЦЕССА ::= АВТОМАТНАЯ-ПРОГРАММА [ || ТЕЛО-ПРОЦЕССА ]
```

В общем случае процесс определен в виде параллельной композиции последовательных процессов, определяемых в виде АВТОМАТНОЙ-ПРОГРАММЫ. Переменная из состояния процесса является разделяемой, если она доступна в различных процессах параллельной композиции.

```
АВТОМАТНАЯ-ПРОГРАММА ::= СЕГМЕНТ [ АВТОМАТНАЯ-ПРОГРАММА ]
```

Автоматная программа является автоматом (машиной конечных состояний) и представлена набором сегментов кода. Вершина автомата - начало сегмента, гипердуга автомата - сегмент. Исполнение сегмента кода завершается передачей управление на начало другого сегмента. Переменные, присваиваемые в одном сегменте и используемые в другом, должны принадлежать состоянию процесса.

```
СЕГМЕНТ ::= МЕТКА : [ inv ИНВАРИАНТ-СЕГМЕНТА ] ТЕЛО-СЕГМЕНТА
ИНВАРИАНТ-СЕГМЕНТА ::= ФОРМУЛА
```

Когда исполнение процесса достигает начала сегмента, соответствующий ИНВАРИАНТ-СЕГМЕНТА, если он имеется, должен быть истинным.

ТЕЛО-СЕКМЕНТА ::= КОМПОНЕНТ-СЕКМЕНТА [ | ТЕЛО-СЕКМЕНТА ]  
 КОМПОНЕНТ-СЕКМЕНТА ::= ОПЕРАТОР | ВЫЗОВ-ПРОЦЕССА

В общем случае сегмент определен в виде недетерминированной композиции компонентов. Результатом исполнения композиции является исполнение одного из компонентов, выбираемого недетерминированно. Исполнение ОПЕРАТОРА всегда завершается либо переходом на начала другого сегмента, либо одним из выходов определяемого процесса.

ВЫЗОВ-ПРОЦЕССА ::= ИДЕНТИФИКАЦИЯ-ПРОЦЕССА ( [АРГУМЕНТЫ] (: РЕЗУЛЬТАТЫ-ВЕТВИ)\* )  
 РЕЗУЛЬТАТЫ-ВЕТВИ ::= [РЕЗУЛЬТАТЫ] [ОПЕРАТОР-ПЕРЕХОДА]

Процесс, запускаемый на исполнение ВЫЗОВОМ-ПРОЦЕССА, должен иметь соответствующее определение процесса. Отсутствие ОПЕРАТОРА-ПЕРЕХОДА по одной из ветвей означает, что исполнение по данной ветви будет продолжено с начала следующего сегмента.

ОПЕРАТОР-ПЕРЕХОДА ::= # МЕТКА  
 ИДЕНТИФИКАЦИЯ-ПРОЦЕССА ::= [ИМЯ-МОДУЛЯ .] ИМЯ-ПРОЦЕССА | [ ОБЪЕКТ-КЛАССА . ] ИМЯ-ПРОЦЕССА | ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ

ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ должно вычислять имя процесса. Если вызываемый процесс является методом некоторого класса, то вызов процесса реализуется через ОБЪЕКТ-КЛАССА.

В теле процесса наряду с операторами, разрешенными в определении предиката, используются дополнительные операторы.

ОПЕРАТОР-РАБОТЫ-С-ПРОЦЕССАМИ ::= ЗАПУСК-НЕЗАВИСИМОГО-ПРОЦЕССА | РЕАКЦИЯ-НА-СООБЩЕНИЯ | ПОСЫЛКА-СООБЩЕНИЯ | ЗАПУСК-ПАРАЛЛЕЛЬНОГО-ПРОЦЕССА | ЗАЩИЩЕННЫЙ-ОПЕРАТОР

Взаимодействие между процессами реализуется посредством приема и передачи сообщений.

ОПИСАНИЕ-СООБЩЕНИЯ ::= **message** ИМЯ-СООБЩЕНИЯ [( СПИСОК-ТИПОВ )]  
 ИМЯ-СООБЩЕНИЯ ::= ИДЕНТИФИКАТОР  
 СПИСОК-ТИПОВ ::= ИЗОБРАЖЕНИЕ-ТИПА [, СПИСОК-ТИПОВ]

Сообщение содержит (возможно, пустой) набор значений, типы которых указываются в параметрах описания сообщения.

ПОСЫЛКА-СООБЩЕНИЯ ::= **send** ИМЯ-СООБЩЕНИЯ [( СПИСОК-ВЫРАЖЕНИЙ )]

Оператор посылает сообщение с набором значений, вычисленных СПИСОКОМ-ВЫРАЖЕНИЙ.

РЕАКЦИЯ-НА-СООБЩЕНИЯ ::= **receive** СООБЩЕНИЕ ОПЕРАТОР | **receive** СООБЩЕНИЕ { ОПЕРАТОР } **else** [ РЕАКЦИЯ-НА-СООБЩЕНИЯ ]  
 СООБЩЕНИЕ ::= ИМЯ-СООБЩЕНИЯ [( СПИСОК-ПЕРЕМЕННЫХ )]

Для второго варианта правила при отсутствии первого сообщения в канале проверяются другие сообщения из **else**-части оператора. Оператор реакции на сообщения исполняется многократно до тех пор, пока в канале не появится одно из сообщений, принимаемых оператором реакции на сообщения. После получения сообщения переменные, указанные СПИСОКОМ-ПЕРЕМЕННЫХ, получают значения параметров сообщения, и выполняется соответствующий ОПЕРАТОР.

Конструкция СООБЩЕНИЕ в любой позиции логического выражения трактуется как неблокирующий прием сообщений. Исполнение конструкции осуществляется следующим образом. В очереди пришедших сообщений исполняемого процесса ищется сообщение с именем, указанным в конструкции СООБЩЕНИЕ. Если сообщение обнаружено, оно вычеркивается из очереди. При этом значения параметров присваиваются соответствующим переменным, перечисленным в круглых скобках. Результатом исполнения конструкции СООБЩЕНИЕ является значение **true**. Если требуемое сообщение не обнаружено в очереди, исполнение конструкции СООБЩЕНИЕ завершается значением **false**.

ЗАПУСК-ПАРАЛЛЕЛЬНОГО-ПРОЦЕССА ::= **spawn** ВЫЗОВ-ПРОЦЕССА

Реализуется запуск процесса, определяемого ВЫЗОВОМ-ПРОЦЕССА, параллельно с процессом, выполняющим данный оператор.



Данная программа может получиться в результате применения первых трех трансформаций: склеивания переменных, замены хвостовой рекурсии циклом и подстановки определения предиката на место вызова.

Циклы `for` и `while` часто используются для представления циклов, реализуемых с помощью операторов перехода, в целях улучшения структуры программы.

## Список литературы

- [1] Шелехов В.И. Введение в предикатное программирование. - Новосибирск, 2002. - 82с. - (Препр. / ИСИ СО РАН; N 100).
- [2] Шелехов В.И. Язык предикатного программирования Р. - Новосибирск, 2002. - 40с. - (Препр. / ИСИ СО РАН; N 101).
- [3] Шелехов В.И. Предикатное программирование: основы, язык, технология. // Методы предикатного программирования / ИСИ СО РАН. - Новосибирск, 2003. - С.7-15.
- [4] Шелехов В.И. Разработка программы построения дерева суффиксов в технологии предикатного программирования. - Новосибирск, 2004. - 52с. - (Препр. / ИСИ СО РАН; N 115).
- [5] Шелехов В.И. Язык спецификации процессов // Методы предикатного программирования. Вып.2 / ИСИ СО РАН. - Новосибирск, 2006. - С. 17-34.
- [6] Шелехов В.И. Модель корректности программ на языке исчисления вычислимых предикатов. - Новосибирск, 2007. - 51с. - (Препр. / ИСИ СО РАН; N 145).
- [7] PVS Specification and Verification System. - <http://pvs.csl.sri.com/>
- [8] Milner R. A Calculus of Communicating Systems. LNCS, Vol. 94. Springer Verlag, 1980.
- [9] Lamport L. Specifying concurrent systems with TLA+. - Calculational System Design, N. 173 in Series F: Computer and Systems Sciences, pages 183-247. - IOS Press, Amsterdam, 1999.
- [10] L. Moura, S. Owre, and N. Shankar. The SAL Language Manual. - SRI International. - <http://sal.csl.sri.com/doc/language-report.pdf>
- [12] Карнаухов Н.С., Першин Д.Ю., Шелехов В.И. Язык предикатного программирования Р — Новосибирск, 2010. — 42с. — (Препр. / ИСИ СО РАН; N 153).

## 12 Дополнения к языку Р

### 12.1 Директивы компилятора

```
ДИРЕКТИВЫ-КОМПИЛЯТОРА ::=
    pragma ДИРЕКТИВА-КОМПИЛЯТОРА (, ДИРЕКТИВА-КОМПИЛЯТОРА)* [БЛОК]
ДИРЕКТИВА-КОМПИЛЯТОРА ::= ИМЯ-ДИРЕКТИВЫ [( ПАРАМЕТРЫ-ДИРЕКТИВЫ-КОМПИЛЯТОРА )]
ПАРАМЕТРЫ-ДИРЕКТИВЫ-КОМПИЛЯТОРА ::=
    ПАРАМЕТР-ДИРЕКТИВЫ-КОМПИЛЯТОРА (, ПАРАМЕТР-ДИРЕКТИВЫ-КОМПИЛЯТОРА)*
ПАРАМЕТР-ДИРЕКТИВЫ-КОМПИЛЯТОРА ::= КЛЮЧЕВОЕ-СЛОВО | ИДЕНТИФИКАТОР | КОНСТАНТА | МЕТКА
```

Директивы могут относиться ко всему файлу, в котором объявлены, к его части, либо к некоторой части исходного кода внутри предиката.

Директивы могут объявляться вне предикатов либо внутри предикатов. Если директивы объявлены вне какого предиката, их действие распространяется с места определения до места переопределения либо до конца файла, если они не переопределяются, при этом БЛОК за их определением должен отсутствовать. Если директивы объявлены внутри предиката, в их определении должен присутствовать БЛОК, на который распространяется их действие.

Определим следующие директивы компилятора: `int_bitness` (РАЗРЯДНОСТЬ-ЦЕЛЫХ), `real_bitness` (РАЗРЯДНОСТЬ-ВЕЩЕСТВЕННЫХ), `int_overflow` (ПЕРЕПОЛНЕНИЕ-ЦЕЛЫХ) и `real_overflow` (ПЕРЕПОЛНЕНИЕ-ВЕЩЕСТВЕННЫХ).

```

ИМЯ-ДИРЕКТИВЫ           ::= int_bitness | real_bitness | int_overflow | real_overflow
РАЗРЯДНОСТЬ-ЦЕЛЫХ       ::= 1 | 2 | ... | 64 | native | unbounded
РАЗРЯДНОСТЬ-ВЕЩЕСТВЕННЫХ ::= 32 | 64 | 128 | unbounded
ПЕРЕПОЛНЕНИЕ-ЦЕЛЫХ      ::= wrap | strict | fail | # МЕТКА
ПЕРЕПОЛНЕНИЕ-ВЕЩЕСТВЕННЫХ ::= safe | strict | fail | # МЕТКА

```

В дальнейшем допускается добавления других директив.

Битности целых и вещественных относятся к типам `int`, `nat` и `real`, объявленным без параметров. По умолчанию они не ограничены.

`int_overflow` и `real_overflow` управляют поведением в случае переполнения. Они могут иметь следующие параметры:

`strict` используется для Семантика цикла `for` соответствует языку C++. Выход из цикла `for` может также быть реализован оператором `break` из тела цикла. означает проверку на переполнение на этапе компиляции, если не удаётся определить, возможно ли переполнение, либо установлен факт возможности переполнения, то возникает ошибка компиляции.

`fail` вызывает проверку времени исполнения, при этом, если происходит переполнение, исполнение останавливается и печатается отладочная информация, которая может помочь программисту выявить и исправить ошибку в программе.

`wrap` означает использование специальных арифметических операций для целых чисел, например, сложение можно определить предикатом, эквивалентным следующему:

```

1 addition (type t, t a, t b : t r) {
2   r = (a + b - low (t)) % (high (t) - low (t)) + low (t);
3 }

```

где на место `high (t)` и `low (t)` во время компиляции подставляются минимальное и максимальное значение типа `t` (будем считать, для определённости, что внутри этой операции вычисления выполняются с произвольной точностью и переполнения не происходит).

`safe` так же означает использование альтернативных операций: если результат операции лежит между минимальным и максимальным значением вещественного типа, он приводится к ближайшему значению, представимому с заданной точностью; если результат больше максимального значения, будем считать его равным `inf` (бесконечности); если же результат меньше минимально допустимого значения, будем считать его равным `-inf`.

При локальном объявлении (т.е., внутри предиката) директивы компилятора `int_overflow` или `real_overflow` допускается использование в качестве параметра метки ветви предиката. В этом случае при переполнении исполнение предиката завершается со срабатыванием соответствующей ветви. Например:

```

1 safe_sum (int (32) a, int (32) b : int (32) result #ok: #err) {
2   pragma int_overflow (#err) { result = a + b; };
3 }
4 // ...
5 safe_sum (a, b : int (32) r #ok: #err)
6   case ok: print ("a + b = ", r)
7   case err: print ("Произошло переполнение!");

```

Предложенный выше подход позволяет не вдаваться в детали реализации там, где этого не требуется, и при этом сохранить полный контроль над машинным представлением чисел в случаях, когда это необходимо.

## 12.2 Использование \* при описании параметров определения предиката

```

ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ ::=
    ИЗОБРАЖЕНИЕ-ТИПА ИМЯ-АРГУМЕНТА (, ИМЯ-АРГУМЕНТА)* |
    ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА
ОПИСАНИЯ-АРГУМЕНТОВ ::= ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ [, ОПИСАНИЯ-АРГУМЕНТОВ]
ИМЯ-АРГУМЕНТА ::= ИДЕНТИФИКАТОР | ИДЕНТИФИКАТОР *

```

Использование аргумента вида `имя*` определяет аргумент `имя` и результат `имя'`. При этом результат `имя'` не должен встречаться в описании результатов предиката. Переменная `имя'` считается находящейся в начале списка результатов предиката. Если имеется несколько аргументов, снабженных звездочкой `*`, то порядок соответствующих результирующих переменных со штрихом `'` в списке результатов соответствует порядку аргументов со звездочкой.

Listing 20: Примеры определений предикатов

```
1 assign (int from : int to) { to = from }
2 main (seq (string) argv : int ret_code) { ret_code = 0 }
3 power (real x*, int p :) { x' = x^p; }
```

```
ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ]
    ТЕЛО-ПРЕДИКАТА [ПОСТУСЛОВИЯ-ВЕТВЕЙ]
ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ИМЯ-ПРЕДИКАТА ( [ОПИСАНИЯ-АРГУМЕНТОВ] : ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ )
```

В ОПИСАНИИ-АРГУМЕНТОВ предиката-гиперфункции не допускается звездочка `*` в конце имени параметра-аргумента.

### 12.3 Предикат комбинирования

Для структурных выражений определён предикат комбинирования, с помощью которой можно сопоставлять значение структуры некоторому произвольному значению определённым образом. Наиболее распространённым примером этому может служить форматирование строк, отображающее строку, в которой задаётся формат, и кортеж в отформатированную заданным образом строку. Рассмотрим пример:

Listing 21: Пример использования `combine`

```
1 type format_value_t = union (int d, real f, string s);
2
3 fmtPart (format_value_t v, string fmt, string s, unsigned int fieldNum : s') {
4     // v:           значение, в виде объединения поддерживаемых типов,
5     // fmt:         строка форматирования,
6     // s:           уже отформатированная часть строки,
7     // fieldNum:   порядковый номер обрабатываемого поля структуры (по этому номеру
8     //             получаем правила форматирования для данного значения)
9 }
10
11 format (string fmt, type structType, structType a : string s) { combine (a, fmt, "", fmtPa
12
13 string date = format ("%02d.%02d.%d", struct (int, int, int), (26, 2, 2008)); // date = "2
```

В общем случае, первым параметром `combine` может быть произвольная структура, вторым — выражение произвольного типа, типы выходного и третьего входного параметра должны совпадать, а четвёртым входным должен быть предикат, со следующими ограничениями на параметры:

- Первый параметр — объединение, такое что для каждого типа, участвующего в кортеже, являющимся первым параметром `combine`, в нём существует единственная альтернатива с таким типом. При этом метка альтернативы может быть произвольной, компилятор выберет альтернативу, основываясь на уникальности типов, а не на метке.
- Второй параметр это в точности второй параметр `combine`. Любое значение, переданное в `combine`, будет передано и в этот предикат.
- Тип третьего параметра должен совпадать с типом третьего параметра `combine`, а так же выходных параметров `combine` и этого предиката. При первом вызове этого предиката значение данного параметра равно значению соответствующего параметра `combine`. При последующих вызовах оно равно результату предыдущего вызова этого предиката.
- Четвёртым параметром передаётся порядковый номер вызова предиката (считая с нуля). Предикат вызывается последовательно для каждого поля входной структуры. Результат последнего вызова возвращается как результат `combine`.

Конечно, предикат комбинирования вводится прежде всего именно чтобы обеспечить возможности реализации форматирования строк, но также очевидно, что для него существуют и другие применения.

## 12.4 Специальная семантика полей типа структуры

При определении полей структуры идентификатор может не указываться, в этом случае доступ к этому полю производится по индексу или с помощью базисных предикатов `head` и `tail`. Если два определения структур отличаются только идентификаторами полей, то они определяют один и тот же тип.

Listing 22: Пример использования полей структур

```

1 type person_t = struct (
2   string familyName,
3   string givenName,
4   nat age
5 );
6 person_t somePerson = ("Smith", "John", 21);
7 person_t someOtherPerson = (givenName: "Jane", familyName: "Doe", age: 18);
8 nat n = somePerson.age;
```

Если при определении константы структурного типа используются идентификаторы полей (см. строку 6 примера), то такая константа совместима также и с теми структурами, у которых совпадают типы и идентификаторы полей, т.е., инициализация в строке 7 примера корректна.

## 12.5 Базисные предикаты с расширенной функциональностью

Таблица 3: Базисные предикаты

	Описание	array	list	map	seq	set	struct
<code>len</code>	Количество элементов	+	+	+	+	+	+
<code>empty</code>	Проверка на пустоту	+	+	+	+	+	+
<code>head</code>	Первый элемент	+	+	-	+	-	+
<code>tail</code>	Элементы, следующие за первым	+	+	-	+	-	+
<code>keys</code>	Множество ключей ассоциативного массива	-	-	+	-	-	-
<code>dim</code>	Число измерений массива (0, 1 или 2)	+	-	-	+	-	-
<code>combine</code>	Комбинирование полей структуры	-	-	-	-	-	+

## 12.6 Дополнительная семантика массивов

Массивы одинакового размера и одинаковым типом элементов совместимы, т.е. выражения одного из этих типов могут неявно приводиться компилятором к другому из этих типов (иначе операторы, вроде определения переменной в строке 4 примера ниже, выглядели бы нелогично из-за возможного несовпадения типов).

Допустимо использование символов “...” вместо описания элементов массива. В этом случае, конкретная размерность массива привязывается к переменной такого типа при инициализации (см. строку 1 примера). Это позволяет использовать одни и те же предикаты для работы с массивами произвольного размера.

Listing 23: Пример операций над массивами

```

type int_mtx_t = array (int, ..., ...);
int_mtx_t m = [ [ 1, 2, 3 ],
               [ 4, 5, 6 ] ];
int size    = len (m); // size = 6
int dim     = dim (m); // dim  = 2
int e      = m [1, 2]; // e    = 6
```

С массивами можно использовать предикат `len`, возвращающий количество элементов, и `dim`, возвращающий количество измерений. Также для них определена операция `+`, возвращающая объединение массивов-операндов (в случае, если их типы элементов совпадают, а множества индексов являются подтипами одного типа и не пересекаются) и логическая операция `in`, возвращающая истину, если её левый операнд является элементом массива (правый операнд), и ложь иначе.

## 12.7 Ассоциативные массивы

ОПРЕДЕЛЕНИЕ-АССОЦИАТИВНОГО-МАССИВА ::= **map** ( ИЗОБРАЖЕНИЕ-ТИПА , ИЗОБРАЖЕНИЕ-ТИПА )  
КОНСТРУКТОР-АССОЦИАТИВНОГО-МАССИВА ::= [ { ( ВЫРАЖЕНИЕ : ВЫРАЖЕНИЕ ) \* } ]

Listing 24: Пример использования ассоциативного массива

```
1 type string_map_t = map (string , string );
2 string_map_t m    = [ { "qwerty" : "asdf", "foo" : "bar" } ];
3 string s          = m [ "foo " ]; // s = "bar"
4 set (string) k    = keys (m); // k = { "qwerty", "foo" }
```

Ассоциативные массивы позволяют ставить в соответствие значения произвольных типов.

Аналогично прочим массивам, для ассоциативных массивов определены операции **len**, **in**, **+**, с той разницей, что **in** проверяет наличие индекса, а не наличие элемента. Также определён предикат **keys**, возвращающий множество индексов.

## Изменения версии 0.11

1. Разд. 10. Аксиомы, леммы и теоремы определяются конструкцией **УТВЕРЖДЕНИЕ**, одной из альтернатив **ОПИСАНИЯ**.
2. Разд. 10. Введена конструкция **ТЕОРИЯ**. Она может встречаться в составе описаний модуля программы. **УТВЕРЖДЕНИЯ**, должны входить в состав некоторой **ТЕОРИИ**, и не может встречаться в программе вне **ТЕОРИИ**.
3. Разд. 10. Откат назад на один шаг. Конструкция **ТЕОРИЯ** устраняется из языка. Конструкция **УТВЕРЖДЕНИЕ** является одной из альтернатив **ОПИСАНИЯ**.

## Изменения версии 0.10

1. Разд. 4. 9. **ОПИСАНИЕ-ПРЕДИКАТА-СПЕЦИФИКАЦИИ** заменено на **ОПИСАНИЕ-ФОРМУЛЫ**. Ранее формула определяла предикат. Сейчас - функцию, тип которой указывается после двоеточия. Если тип - **bool**, то формула определяет предикат.
2. Разд. 5. Из правила **ОПРЕДЕЛЕНИЕ-ЛОКАЛЬНОГО-ПРЕДИКАТА** изъято (удалено) **ОПРЕДЕЛЕНИЕ-СУЖЕННОГО-ПРЕДИКАТА**. Эта форма оказалась избыточной.
3. Разд. 6. В конструкцию **МОДИФИКАЦИЯ** добавлен разделитель **with**.
4. Разд. 4. Введены формальные параметры модулей и фактические параметры при импорте модулей. Импорт может оказаться не в начале модуля.
5. Из агрегатов исключены объединения.
6. Разд. 7. Тип объединения **union** теперь имеет структуру алгебраического типа. Альтернатива объединения - конструктор с аргументами, называемыми полями объединения. Тип перечисления определен как частный случай типа объединения.
7. Термин “последовательность” везде заменяется на “список”. Описатель **seq** заменяется на **list**. Вместо функций **head** и **tall** используются поля **car** и **cdr**. Материал из дополнений к языку, разд. 12.7, определяющий списки, частично переносится в разд. 6 и 7. Вводится конструкция **АГРЕГАТ-СПИСОК**. Пустой агрегат теперь не представляется константой **nil**.
8. Разд. 7. Удалено **ИЗОБРАЖЕНИЕ-ТИПА-С-ПУСТЫМ-ЗНАЧЕНИЕМ**.
9. Разд. 6. Префиксом для элемента массива, поля структуры и поля класса является **ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ**, а не **ВЫРАЖЕНИЕ**.
10. Разд. 5. Удалено предложение: Оператор выбора должен быть полным: при отсутствии части **default** набор значений кодов должен покрывать набор возможных значений выражения в заголовке. — есть случаи, когда оно может не выполняться
11. Разд. 7. Исправлены примеры программ со списками (листинги 7, 8, 10, 13)
12. Разд. 7. Удалено **ИЗОБРАЖЕНИЕ-ТИПА-С-ПУСТЫМ-ЗНАЧЕНИЕМ**

13. Разд. 11. Подраздел “Директивы компилятора” пока перемещен в Дополнения к языку. Эта часть требует серьезной доработки, и сейчас в публикации ее лучше не показывать.
14. Переставлены разделы 9 и 10.
15. Разд. 11. Удален описатель `mutable`, поскольку пользователь не может программировать непосредственно на императивном расширении.
16. Разд. 4. Изменен синтаксис конструкции ИМПОРТ-МОДУЛЯ
17. Разд. 5. Изменен синтаксис. Описания переменных могут находиться перед любым оператором. Сделаны уточнения синтаксиса условного и параллельного операторов.
18. Разд. 4. Вместо `printf` языка C++ используется собственный оператор `print`
19. Разд. 7, 8.3. Тип может отсутствовать при описании переменных-полей конструктора в альтернативе `case` оператора `switch` и переменных - индексов массива в операторе `for`.

## Изменения версии 0.9

1. Вместо оператора КОНСТРУКТОР-МАССИВА теперь используется операция ОПРЕДЕЛЕНИЕ-МАССИВА.
2. Вместо `lambda` используется `predicate`.
3. Разд. 8.4. Описана семантика операции объединения массивов.
4. Разд. 6. Упрощена конструкция АГРЕГАТА. Вместо ПРИСОЕДИНЕННОГО-АГРЕГАТА используется операция МОДИФИКАЦИЯ.
5. Введены конструкции ОПИСАНИЕ-ПРЕДИКАТА и ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА. Упрощается определение генератора предиката и изображения предикатного типа.
6. Разд. 9. Изменения в определении ФОРМУЛЫ. Введено описание предиката спецификации. Введены импликация и логическое тождество.
7. Разд. 9. Добавлен ограничитель `formula` при описании предиката спецификации. Вместо `<=>` используется просто `=`
8. Разд. 5. В одной альтернативе оператора выбора допускается несколько кодов альтернатив.
9. Разд. 8.2, 8.3. Исправлен синтаксис ИНДЕКСА-ЧАСТИ и др. связанных понятий
10. Разд. 9. Обобщено понятие СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ
11. Разд. 10. В защищенном операторе изменен синтаксис
12. Разд. 11. В императивном расширении удален групповой оператор присваивания, поскольку это частный случай оператора присваивания.
13. Разд. 7. Уточнение семантики: изображение диапазона является подмножеством типа `int`
14. Разд. 7. Введены операции `last` и `pred` для последовательностей.
15. Разд. 4, 8.3. Для описания переменной возможно использование описателя `var`.
16. Разд. 9. Кванторы могут также изображаться ограничителями `forall` и `exists`.
17. Разд. 6. Для мультипеременной и мультивыражения допускается опускать скобки “|”
18. Разд. 3.3. Описатель типа локала в позиции результата вызова относится только к этому локалу и не распространяется на последующие результаты.
19. Разд. 5. В операторной позиции может находиться определение локального предиката. Частным случаем является суженный предикат.
20. Разд. 7. Вместо `pred` теперь используется `prec`.
21. Разд. 7. Базисными типами при изображении диапазона могут быть `int`, `enum`, или `char`